

BASIC

PARA
MICROCOMPUTADORES

THEM

RETURN

10

INPUT

PRINT

IF GOTO

END

LIST

M A N U A L
D E
B A S I C
P A R A C O M P U T A D O R E S

R. MIRANDA

M. PRENDEZ

M. MIRANDA

Inscripción N° 59.214.

Diseño de la Portada:
Srta. Pamela Miranda.

Edición N° 2 - Septiembre 1984 Santiago

Aprobado por el Ministerio de Educación

Editado en Azócar Multicopias Ltda. Moneda 1141

I N D I C E

PRIMERA PARTE

PAGINA

1.	Introducción	3
2.	Descripción del Computador	4-5
3.	Examen y práctica del teclado	6-7-8
4.	Uso de la tecla CONTROL [CTRL]. 1a. parte	9
5.	PRINT. Modo directo y modo de programación	10-13
6.	Operadores: dos puntos (:). coma (,), punto y coma (;)	14
7.	Escritura de un programa	15
8.	Comando LIST o L., PRINT	16-17
9.	Comando NEW	17
10.	Instrucción Remark (REM), END, STOP	18
11.	Instrucción POSITION (POS.)	19
12.	Modo de Contar del Computador	20-21
13.	El computador y los números	22-23
14.	Mensajes de Error	24
15.	Uso de la tecla CONTROL [CTRL]. 2a. parte	25-26
16.	Variables. Numéricas y alfanuméricas. Conceptos	27-28
17.	Instrucción LET. DIM.	29
18.	Instrucción INPUT.	30
19.	Instrucción GOTO. Contadores.	31-32
20.	Instrucción ON/GOTO. Menus.	33
21.	Instrucción FOR/NEXT (loop o bucle)	34-36
22.	Instrucción IF/THEN	37-39
23.	Instrucción GOSUB/RETURN	40
24.	Instrucción ON/GOSUB	41
25.	Instrucción READ/DATA	42-43
26.	Instrucción RESTORE	44
27.	ARRAYS (arreglos)	45-46
28.	MATRIX (matrices)	47-48

SEGUNDA PARTE

29.	Grabadora de Programas	1-5
30.	Introducción a Gráficos	6-7
31.	Instrucción SETCOLOR (Juego o Grupo de Colores)	8-9
32.	Instrucción Color (Color)	10-11
33.	Instrucción PLOT	12

(Continuación Índice)

PAGINA

34.	Instrucción DRAWTO X,Y	12
35.	Modo de Gráficos 0.	13-15
36.	Modo de Gráficos 1.	16-17
37.	Modo de Gráficos 2.	18
38.	Modo de Gráficos 3, 5 y 7	19-20
39.	Modo de Gráficos 4 y 6	21
40.	Modo de Gráficos 8	22
41.	Resumen de SETCOLOR y COLOR	23
42.	Gráficos GTIA. Modos de Gráficos 9, 10 y 11	24-25
43.	Modos de Gráficos 12 y 13	25
44.	Modos de Gráficos 14 y 15.	26
45.	Sonido	27-28
46.	Valores de Sonido para las Notas Musicales	29
47.	Ejercicio	30
48.	Instrucción PEEK e Instrucción POKE	31-32

APENDICES.

TABLA RESUMEN DE MODOS GRAFICOS	i
CODIGOS DE ERROR	ii-iii
LISTADO DE PALABRAS RESERVADAS	IV
PLANTILLA DE CARACTERES GRAFICOS	tapa posterior

PRIMERA PARTE

INTRODUCCION

Nuestra intención al preparar estos apuntes, es proporcionar a los dueños de computadores ATARI XL un texto de estudio y de consulta, para manejar en forma ágil un conjunto básico de comandos e instrucciones que permitan iniciarse en la programación de microcomputadores que acepten el lenguaje BASIC (Beginners All-purpose System Information Code).

Este manual tomará como modelo la línea de microcomputadores ATARI XL y nos referiremos a ella cuando se trate el tema del teclado. El BASIC que trataremos será, principalmente, el ATARI-BASIC.


Simbología en escritura:

[RETURN]	= presione tecla RETURN.
[CTRL] [INSERT/>]	= presione la tecla CTRL y manténgala oprimida mientras presione la tecla INSERT/>.
[ESC] - [CTRL] [CLEAR/<]	= presione la tecla ESC una vez y suéltela. Luego presione CTRL y manténgala oprimida mientras presiona CLEAR/<.

DESCRIPCION DEL COMPUTADOR

Su computador ATARI se divide en varias partes o segmentos. Algunos de ellos pueden usarse a través de comandos especiales, y otros sólo son usados por el fabricante.

Examine la figura y analice las partes que componen su computador.



Los nombres de los dispositivos a los que Ud. tiene acceso son:

- | | |
|---------------------------------|-------------------------------|
| - Grabadora de Cintas | (Cassette) : "C:" |
| - Grabadora de Discos flexibles | (Diskettes): "D:" (D1: a D4:) |
| - Pantalla del televisor | (Screen) : "S:" |
| - Editor de programas | (Editor) : "E:" |
| - Teclado alfanumérico | (Keyboard) : "K:" |
| - Impresora de papel | (Printer) : "P:" |
| - Interfase para modems y otros | (seRial) : "R:" (R1 a R4) |

Los nombres de los dispositivos a los que Ud. tiene acceso son:

- Grabadora de Cintas (Cassette) : "C:"
- Grabadora de Discos flexibles (Diskettes): "D:" (D1: a D4:)
- Pantalla del televisor (Screen) "S:"
- Editor de programas (Editor) : "E:"
- Teclado alfanumérico (Keyboard) : "K:"
- Impresora de papel (Printer) : "P:"
- Interfase para modems y otros (seRial) . "R:" (R1 a R4)

Tenga presente que es importante escribir las letras claves entre comillas y con los dos puntos al final. En ciertos casos, después de los dos puntos, se puede poner un nombre y luego cerrar comillas.

EXAMEN Y PRACTICA DEL TECLADO

a) Consejos y sugerencias para el programador principiante:

NO TENGA TEMOR! Nada de lo que Ud. escriba o pueda escribir dañará al computador o a su televisor.

Cometer ERRORES forma parte del aprendizaje. Todo el mundo los comete, de modo que acostúmbrese a la idea.

En general, los computadores usan letras mayúsculas y no aceptan instrucciones escritas con letras minúsculas o símbolos gráficos.

Cuando, al escribir una instrucción, llegue al final de la línea, continúe escribiendo como si estuviera al principio o al medio de la línea. El computador se encargará de volver el cursor al principio de la línea siguiente.

Después de cada instrucción o de cada operación que Ud. escriba en pantalla, es necesario que el computador "tome nota de ella" o "se dé cuenta que Ud. la escribió". No olvide, por lo tanto, presionar la tecla ENTER o RETURN para ingresar la instrucción, una vez escrita.

Además, debe tener en cuenta que el computador no se equivoca jamás. Sólo se equivocan los programadores. Uno de los errores más comunes es confundir la letra O con el número Ø. Usted no debe equivocarse ni confundirse. El computador no lo hará. Si Ud. escribe una letra O, aparecerá en la pantalla un símbolo que es distinto, aunque parecido, al que Ud. verá si presiona la tecla correspondiente al número Ø. Observe bien la diferencia y trate de no equivocarse.

b) Una mirada al teclado:

Aunque el teclado de un computador es muy similar al de una máquina de escribir, hay algunas características especiales que Ud. debe conocer cuanto antes, y otras que veremos más adelante.

Antes de continuar con la próxima lección, practique con las siguientes teclas:

[DELETE/BACK S]

Esta tecla borrará cualquier carácter que Ud. escriba por error. Escriba algunas letras y luego "BACKSPACE" sobre ellas para captar la idea.

Existen algunas acciones que deben ser ejecutadas mediante la combinación de dos teclas. Se requiere mantener una tecla presionada mientras se oprime la otra. La primera es la combinación necesaria para imprimir "comillas". Presione la tecla [SHIFT] y, manteniéndola oprimida, oprima la tecla ["/2]. ([SHIFT]["/2]).

[SHIFT][CLEAR/<]

Esta tecla le ayudará si Ud. tiene problemas con lo que ocurre en la pantalla. Hasta que se familiarice con el teclado, la pantalla podría llenársele de mensajes de error y/o caracteres que Ud. no entiende. Para limpiar la pantalla y empezar de nuevo, presione [SHIFT] y sin soltarla, oprima [CLEAR/<].

[RETURN]

([ENTER] en algunos computadores). Al presionar esta tecla Ud. le indica al computador que ha ingresado alguna información o respuesta.

NADA SUCEDE A MENOS QUE UD. PRESIONE RETURN!!!

[/ | \]

La tecla [símbolo ATARI] lo pone a Ud. frente al video inverso. Oprímala y escriba algo. Oprima de nuevo la tecla y volverá al modo de presentación normal.

[CAPS/LOWR]

La tecla [CAPS/LOWR] le dará letras minúsculas en la pantalla. Presiónela y escriba algunas letras. Para volver a las mayúsculas, oprima [SHIFT][CAPS/LOWR].

Experimente con la tecla ATARI y la de minúsculas, separadamente.

Las características de VIDEO INVERSO y MINUSCULAS son interesantes y útiles, pero para aprender es aconsejable trabajar en modo de presentación NORMAL y con letras MAYUSCULAS.

Cuando su computador ATARI muestra en pantalla la palabra READY, significa que está listo para aceptar sus instrucciones.

Junto a la palabra READY aparece un cuadrado blanco que le indica el lugar donde empezarán a escribirse las próximas palabras.

Muy pronto se dará cuenta que para programar en BASIC, o en cualquier otro lenguaje, todo consiste en saber cuáles son las órdenes que debe darle a su computador y cómo debe hacerlo.

Nota: En la línea ATARI XL la tecla con el símbolo ATARI está reemplazada por una con un rectángulo bicolor.

USO DE LA TECLA CONTROL [CTRL]. 1a. parte.

El aprendizaje y dominio del uso de esta tecla, le permitirán un completo control sobre la pantalla de su computador.

Al utilizar [CTRL] en combinación con otra tecla, debe mantenerla oprimida mientras presiona la otra tecla.

Por ejemplo, presione [CTRL] y, manteniéndola presionada, oprima ["/2]. Observe lo que ocurre. Ensaye con otras teclas y tome nota de lo que sucede.

Oprima [CTRL], presione [_ / -] y manténgalas.
Oprima [CTRL], presione [T / =] y manténgalas.
Oprima [CTRL], presione [\ / +] y manténgalas.
Oprima [CTRL], presione [^ / *] y manténgalas.

El cursor se ha desplazado en forma continua hacia arriba, abajo, izquierda y derecha. Este desplazamiento se hace sobre el texto que se encuentra en pantalla, sin borrarlo.

Con la tecla [CTRL] oprimida y utilizando las teclas con flechas, se puede desplazar el cursor a cualquier lugar de la pantalla, sin interferir con lo que aparece en ella.

Más adelante, veremos otros usos de la tecla [CTRL].

A estas alturas, Ud. habrá notado que si deja las teclas oprimidas por más de un segundo, éstas se repiten ininterrumpidamente. Si no lo ha notado, ensáyelo ahora. Esto es muy útil para desplazar el cursor en forma rápida. Asimismo, verifique lo que sucede cuando el cursor llega al extremo de la pantalla.

PRINT. Modo directo y modo de programación.
(PRINT = IMPRIMIR)

Uno de los comandos que Ud. utilizará con mayor frecuencia, se logra escribiendo las letras

P R I N T

Este conjunto de letras le indica al computador que debe presentar algo en pantalla.

Cuando Ud. escribe PRINT y oprime [RETURN] en seguida, le está diciendo al computador que imprima, o presente en la pantalla, UNA LINEA EN BLANCO.

Escriba lo siguiente:

a) PRINT "HOLA, YO SOY SU COMPUTADOR ATARI." [RETURN]

Cuando llegue al final de la línea siga escribiendo. El computador se encargará del cambio de línea.

Listo? La pantalla debiera mostrar:

READY

PRINT "HOLA, YO SOY SU COMPUTADOR ATARI."

HOLA, YO SOY SU COMPUTADOR ATARI.

READY

El computador le ha obedecido. Ha presentado en la pantalla la frase que Ud. le pidió que mostrara.

Ahora escriba:

b) PRINT 5+3 [RETURN]

c) PRINT 12-8 [RETURN]

En pantalla debiera ver:

READY

PRINT 5+3

8

READY

PRINT 12-8

4

READY

Escriba:

d) PRINT 25*25 [RETURN] (Nota: * es signo de multiplicación)

e) PRINT 240/60 [RETURN] (Nota: / es signo de división)

Qué tenemos en pantalla?

En la misma forma anterior, el computador presentará los resultados 625 y 4.

f) PRINT "2+2" [RETURN]

g) PRINT "2+2 SON ";2+2 [RETURN]

h) PRINT "YO SIGO UN CURSO DE BASIC" [RETURN]

Observe la pantalla y compare los resultados obtenidos. Puede sacar conclusiones generales?

Nota: PRINT puede ser abreviado con el signo "?". Sin embargo, para los principiantes resultará más claro escribir la palabra completa.

Si Ud. le ordena al computador: ? 2+2, lo que verá en la pantalla será:

? 2+2

4

READY

Para recordarlo mejor, imagínese que Ud. le pregunta al computador: Cuánto es 2+2?

Conclusiones.

El computador se comporta como una calculadora que, en lugar de utilizar la tecla [=] para efectuar una operación, utiliza la tecla [RETURN]. Además, debe usarse la instrucción PRINT.

Si, después de escribir PRINT y abrir comillas, Ud. escribe una operación aritmética, una frase o una combinación de letras, signos o números, y luego cierra las comillas, el computador lo interpretará como que Ud. desea reproducir en la pantalla los caracteres que están incluidos entre las comillas. En este caso, no ejecutará las operaciones, si las hubiere.

Este modo de operar que acabamos de describir, se llama MODO DIRECTO. Este modo ejecuta las instrucciones inmediatamente después de que Ud. oprime [RETURN], y es muy útil cuando se precisan respuestas rápidas como las de la calculadora, o para experimentar y ver como responde el computador ante instrucciones determinadas.

Todo claro hasta ahora?

Sigamos entonces!

Escriba:

```
10 PRINT "HOLA, YO SOY SU COMPUTADOR ATARI." [RETURN]
```

Observe que hay un número de línea delante del comando PRINT, y que esta vez el computador no hizo caso de su orden.

Escriba: RUN [RETURN]

La pantalla debe verse así:

READY

```
10 PRINT "HOLA, YO SOY SU COMPUTADOR A  
TARI."
```

RUN

```
HOLA, YO SOY SU COMPUTADOR ATARI.
```

READY

Qué diferencia observa entre este modo de trabajar y el anterior?

En el MODO DIRECTO, no hay un número delante del comando PRINT, y la instrucción se ejecuta inmediatamente al oprimir la tecla [RETURN], después de haber escrito la frase.

En este modo, la instrucción se ejecuta después de que Ud. escribe el comando RUN. Esta forma de trabajo se llama MODO DE PROGRAMACION, y la línea 10 es una línea de PROGRAMA. Este es un programa de una sola línea.

A continuación, experimente con estos nuevos programas, teniendo cuidado de escribirlos de la forma como aparecen en el texto, con el mismo número de espacios.

Programa 1

```
10 PRINT "(aquí escriba su nombre....)"
20 GOTO 10
```

(para detener el programa oprima [BREAK])

Programa 2

```
10 PRINT "ATARI"
20 GOTO 10
```

(para detenerlo oprima [BREAK])

Programa 3

```
10 PRINT "ATARI",
20 GOTO 10
```

(ponga más de 1 coma y experimente)
(para detenerlo oprima [BREAK])

Programa 4

```
10 PRINT "ATARI";
20 GOTO 10
```

(para detenerlo oprima [BREAK])

Programa 5

```
10 PRINT "ATARI": GOTO 10
```

Programa 6

```
10 PRINT "ATARI";: GOTO 10
```

Programa 7

```
10 PRINT "COMPUTADOR": PRINT "ATARI": GOTO 10
```

Programa 8

```
10 PRINT "COMPUTADOR";: PRINT "ATARI";: PRINT "EL MEJOR";
20 GOTO 10
```

Programa 9

```
10 PRINT "COMPUTADOR"; "ATARI"; PRINT "EL MEJOR": GOTO 10
```

Programa 10

10 Invente por lo menos 5 programas para practicar PRINT y los operadores : ; , y lea la lección siguiente.

OPERADORES

DOS PUNTOS.

Usted puede usar los dos puntos para incluir más de una instrucción en una línea de programa. Esto le permite:

- ahorrar tiempo
- ahorrar memoria
- organizar mejor su programa

LA COMA.

La coma le permite presentar la información deseada en forma de columnas. Al poner (,) entre los ítems a escribir, el computador pondrá las palabras o números separados por 10 espacios.

```
10 PRINT "ATARI", "SINCLAIR", "OSBORNE", "MICROPROFESSOR",  
  "CASIO"  
20 PRINT "1", "2", "3", "4", "5", "6"
```

PUNTO Y COMA.

Este operador permite poner las cosas juntas, una al lado de la otra.

```
10 PRINT "HOLA": GOTO 10
```

NEW

```
10 PRINT "HOLA";: GOTO 10
```

NEW

```
10 PRINT "PRECIO DEL DOLAR =" ; "$ 80"
```


ESCRITURA DE UN PROGRAMA

Como ya es obvio, los programas se caracterizan por estar constituidos por líneas de programa, cada una con su número, formando una secuencia ordenada de instrucciones.

A menos que se le instruya de manera diferente, el computador SIEMPRE empezará a ejecutar el programa partiendo por la línea con el menor número relativo, siguiendo hasta la de mayor número.

Al escribir un programa, es una buena idea el asignar números de 10 en 10 a las líneas del programa. De este modo, Ud. deja números disponibles para intercalar nuevas líneas entre las existentes, lo que resulta necesario en la mayoría de los casos.

REEMPLAZO DE LINEAS. La manera más fácil de reemplazar o cambiar una línea es escribirla de nuevo. La línea nueva reemplaza a la antigua cuando la nueva línea sea más corta que la anterior. Si Ud. mantiene el mismo número de línea en ambos casos, la segunda línea reemplaza a la primera.

BORRADO DE LINEAS. Si Ud. quiere hacer desaparecer una determinada línea de programa, escriba el número de la línea y oprima [RETURN].

Ej.: 15 [RETURN]

La línea 15 ya no existe en su programa.

MAXIMO TRES LINEAS. Al escribir programas, necesitará una nueva línea de programa por cada 114 caracteres mostrados en la pantalla, es decir, cada 3 líneas.

Una línea de programa también se llama una línea lógica. Por lo tanto una línea lógica puede tener hasta tres líneas físicas.

NO OLVIDAR!!!

Al escribir programas, DEBE oprimir [RETURN] al término de una línea de programa. De este modo, el computador la considerará ingresada y la tomará en cuenta. Cada vez que quiera "entrar" una instrucción, ya sea en modo directo o en modo de programación, debe hacerlo mediante la tecla [RETURN] o [].

En adelante, omitiremos el indicar explícitamente la tecla [RETURN] al escribir líneas de programa o instrucciones directas al computador, pero Ud. DEBE hacerlo de todos modos.

COMANDO LIST o L., LPRINT (LISTA - LISTEE)

LIST (abreviado L.) es un comando que hace que el computador presente en la pantalla el listado completo del programa.

Escriba:

```
10 FOR K=1 TO 5
20 PRINT "YO VIVO EN SANTIAGO DE CHILE"
30 NEXT K
40 END
```

Corra el programa. Para hacerlo, escriba RUN. En pantalla debe aparecer la frase de la línea 20, sin las comillas, 5 veces.

Ahora escriba LIST.

En la pantalla debe aparecer el listado del programa que acaba de escribir, es decir, las 4 instrucciones que lo componen (líneas 10, 20, 30 y 40).

LISTADO PARCIAL DE UN PROGRAMA.

Cuando empiece a trabajar con programas más largos, estos no cabrán enteros dentro del área que cubre la pantalla y, es posible que la o las líneas que a Ud. le interesa revisar, queden fuera de la pantalla. En tal caso, puede seleccionar las líneas que desea que el computador le muestre.

Para ello debe escribir:

```
LIST XX -si desea ver la línea XX en la pantalla, o
LIST XX,YY -si desea ver desde la línea XX hasta la YY.
                NO OLVIDE USAR LA COMA.
```

Si Ud. todavía no tiene impresora, pase a la próxima lección.

Para sacar el listado de un programa a la impresora o printer, escriba:

LIST"P:

LPRINT es un comando para operar la impresora y obtener una copia "dura", o impresa en papel, de los caracteres que siguen al comando LPRINT, en la línea de programa que se trate.

Por ejemplo, escriba:

```
10 FOR K=1 TO 5
20 LPRINT "YO VIVO EN SANTIAGO DE CHILE"
30 NEXT K
40 END
```

El printer imprimirá 5 veces la frase de la línea 20, sin las comillas.

NOTA: No deberá usar el comando LPRINT sin disponer de un printer. Si lo intenta, obtendrá un mensaje de error, el ERROR 138, que significa que Ud. intentó imprimir sin printer.

COMANDO NEW (NUEVO)

NEW borrará su programa de la memoria del computador. No hay segunda oportunidad.

Si no tiene ningún programa en memoria, escriba uno corto. Listéelo y en seguida escriba NEW. Vuelva a escribir LIST y verá que su programa ha desaparecido. En la pantalla sólo verá:

READY
NEW

READY
LIST

READY

Siempre piénselo dos veces antes de escribir NEW. Si el programa es bueno, puede desear guardarlo impreso en papel (impresora), en cinta (grabadora) o en diskette (unidad de disco). Si no cuenta con estos accesorios, aun puede querer copiarlo de la pantalla a mano.

NOTA: Apagar el computador o abrir el compartimento del cartridge, también borrará su programa. Sin embargo, el apagar el televisor no lo afectará.

INSTRUCCION REMARK, END, STOP. (COMENTARIO, FIN, PARE)

La instrucción Remark (REM), le sirve como guía para poder entender un programa.

Si Ud. escribe un número de línea y a continuación escribe REMARK o REM, puede escribir hasta 3 líneas de información que no afectarán en nada la ejecución del programa.

Cuando el computador "lee" una instrucción REM, no hace nada y salta a la siguiente instrucción. Por consiguiente las instrucciones de REM, deben estar aisladas, y no seguidas por otras que tengan importancia para el desarrollo del programa, escritas en la misma línea lógica de programa.

END es una instrucción que le indica al computador que el programa ha terminado. No siempre es necesario usarla, sobre todo en programas cortos, o que no hacen uso de instrucciones del tipo GOTO o GOSUB. Sin embargo, hay que estar consciente de su existencia, para utilizarla cuando sea necesario.

STOP es una instrucción que le permite detener el programa (por ejemplo para hacer una tarea antes de seguir), sin afectar el desarrollo del mismo. Para reanudar la ejecución, escriba CONT. y oprima [RETURN].

INSTRUCCION POSITION o POS. (POSICION)

POSITION le indica al computador en que lugar de la pantalla debe empezar a escribir. Para ello, la pantalla está dividida en 40 bloques en el sentido horizontal (x) y 24 en el sentido vertical (y). Ello define un sistema de coordenadas en el que el bloque $x=0$, $y=0$ está en la esquina superior izquierda de la pantalla.

Si Ud. no utiliza la instrucción POS. antes de una instrucción PRINT, el computador automáticamente saltará 2 espacios hacia la derecha y empezará a escribir, de izquierda a derecha, en la línea siguiente a donde estaba antes de encontrarse con la instrucción PRINT.

La instrucción POS. se maneja así:

POS. X,Y

X= número de espacios hacia la derecha.
Y= número de espacios hacia abajo.

EJEMPLO:

Recuerde: si hay algún programa en la memoria, debe borrarlo antes de escribir el siguiente. Para ello, escriba NEW.

```
10 REM ESTE PROGRAMA MUESTRA EL USO
  DE LA INSTRUCCION POSITION.
20 POS. 0,2
25 PRINT "ARRIBA A LA IZQUIERDA"
30 POS. 14,13
35 PRINT "EN EL MEDIO"
40 POS. 22,23
45 PRINT "ABAJO A LA DERECHA"
50 GOTO 50
```


ATENCION: Si tiene la pantalla escrita con algunos caracteres, oprima SYSTEM RESET para limpiarla. Más adelante aprenderá otros métodos más eficientes para hacer desaparecer caracteres indeseados de la pantalla.

MODO DE CONTAR DEL COMPUTADOR

Por razones técnicas, los sistemas de numeración internos del computador utilizan el 0 como primer número, en lugar del 1.

Por ejemplo, para contar 10 elementos, el computador los llama: 0, 1, 2, 3, 4, 5, 6, 7, 8 y 9.

Al emplear la instrucción POS., los 40 bloques en el sentido horizontal son designados 0, 1, 2, hasta 39, y los 24 bloques hacia abajo son 0, 1, 2, hasta 23.



PANTALLA

El computador ATARI tiene 16 modos de gráficos, que se dividen en 5 modos para texto y 11 modos para gráficos propiamente tales. Vea el apéndice 1.

Estos modos se invocan mediante el comando GRAPHICS N (N es cualquier número de 0 a 15).

Para utilizar la instrucción POS., debe tenerse en cuenta el modo gráfico en que se encuentra, de otro modo se obtendrá un ERROR 141 - CURSOR FUERA DE PANTALLA.

Utilice el siguiente ejemplo para ubicar un punto en el centro de la pantalla, en cualquiera de los modos indicados.

En gráficos 0 (texto normal)

```
10 GR.0
20 POS. 19,12
30 PRINT "A"
40 GOTO 40
```


En gráficos 2 (texto, letras grandes)

```
10 GR.2  
20 POS. 9,6  
30 PRINT #6; "A"  
40 GOTO 40
```

En gráficos 8 (gráficos de alta resolución)

```
10 GR.8  
20 POS. 159,96  
30 PRINT #6; "i"  
40 GOTO 40
```


EL COMPUTADOR Y LOS NUMEROS

a) Notación para escribir números muy grandes o muy pequeños.

Pueden tener valores enteros o valores con coma flotante. Los valores numéricos están restringidos al rango de -10^{97} a $+10^{97}$.

El computador ATARI maneja números que tengan hasta 9 cifras significativas en la pantalla. Para trabajar con números más grandes utiliza la notación científica.

Escriba: PRINT 123456789012

En la pantalla verá: 1.23456789 E+11

Lo que significa: 1.23456789 por 10 elevado a +11, o sea,
 $1.23456789 * 100.000.000.000$

El computador ATARI, en notación normal, es capaz de manejar números hasta de 10 dígitos.

Números menores que 1.

Escriba: PRINT 0.1, 0.01, 0.001, 0.0001

En la pantalla verá: 0.1 0.01 1.0 E-03 1.0 E-04

Donde 1.0 E-04 significa: 1 por 10 elevado a menos 4, o sea,
 1 dividido por 10.000

Cuando el valor de un número con coma flotante se acerca a cero, o sea, es más chico que $+ 9.99999999 * 10^{-98}$ (9.99999999 dividido por un 1 seguido de 98 ceros), ATARI BASIC lo convierte en cero.

b) Orden de prioridad en operaciones matemáticas.

El computador calcula las expresiones matemáticas en un cierto orden. Los cálculos del mismo tipo o de igual prioridad, los ejecuta de izquierda a derecha.

SIMBOLO	DESCRIPCION
()	- cualquier cosa dentro del PARENTESIS, se ejecuta primero, y luego,
^	- la ELEVACION A POTENCIA, como $5 \wedge 3$, que significa $5*5*5$ ó 125, y después,
* ,/	- MULTIPLICACION Y DIVISION, ejecutadas

en orden de aparición y de izquierda a derecha, y por último,
 + , - - SUMAS Y RESTAS, ejecutadas en orden de precedencia, de izquierda a derecha.

Ejemplos. Vea la diferencia entre:

$$\begin{aligned} 3 + 4 * 5 &= 23 \\ (3 + 4) * 5 &= 35 \\ 5 - 4 / 2 &= 3 \\ (5 - 4) / 2 &= 0.5 \end{aligned}$$

En caso de duda, ponga entre paréntesis lo que desea que se ejecute primero.

Además de las operaciones aritméticas indicadas en la lección sobre el Modo Directo, y las anteriores, su computador tiene una gran capacidad matemática, que puede emplearse en Modo Directo (como calculadora) o en Modo de Programación (para resolver cualquier ecuación que su programa requiera).

Su computador eleva a potencia, (A); obtiene raíz cuadrada, SQR(N); funciones trigonométricas en radianes (normalmente) o en grados (DEG). Además de facilitar la solución de ecuaciones, estas funciones le permitirán hacer entretenidos dibujos, como verá más adelante (lecciones de gráficos).

PRACTICA:

```
10 PRINT "ESCRIBA UN NUMERO ENTRE 1 Y 100";: INPUT A
20 GR.0 : DEG
30 PRINT "SU NUMERO ES:";A
40 PRINT "12 POR ";A;" = ";12*A
50 PRINT "100 DIVIDIDO POR ";A;" = ";100/A
60 PRINT A;" ELEVADO AL CUBO = ";A^3
70 PRINT "RAIZ DE ";A;" = ";SQR(A)
80 PRINT "LOGARITMO DE ";A;" = ";LOG(A)
90 PRINT "COSENO DE ";A;" GRADOS = ";COS(A)
100 PRINT "SENO DE ";A;" GRADOS = ";SIN(A)
110 PRINT "PROGRAMA ...TERMINADO":END
```


MENSAJES DE ERROR

(Ver apéndice 2)

El computador ATARI le da dos tipos de mensajes de error.

ERRORES DE SINTAXIS

Si Ud. escribe una línea sin la puntuación adecuada, sin los operadores correctos, o con cualquier cosa que el computador no entienda, el computador le mostrará la palabra ERROR en la siguiente línea de la pantalla, a continuación de la última línea que Ud. "entró".

Ejemplos:

PRIMT "HOLA" dará por resultado

ERROR - PRIMT "HOLA"

Por inspección se advierte una M en vez de N en PRINT.

print "hola" resulta en

ERROR- print "hola"

Instrucciones no son aceptadas en minúsculas.

Debe usar mayúsculas.

"PRINT HOLA" se convierte en

ERROR - "PRINT "HOLA"

Las comillas deben abrirse después de PRINT y no antes.

ERRORES DE LOGICA

Si Ud. ha escrito las instrucciones correctamente desde el punto de vista de la sintaxis y, al correr el programa, el computador se encuentra con que el desarrollo es ilógico, detendrá la ejecución y le dará un mensaje de error especificando un número y haciendo referencia a la línea en la que se encuentra dicho error.

```
10 LPRINT "DESCRIPCION ARTICULOS"
```

```
RUN
```

```
ERROR 138 AT LINE 10
```

Primero observe la línea 10 por si hay errores obvios. No los hay.

Seguidamente, observe el número del error que se señala. Consulte la lista de mensajes de ERROR, que hay en el Manual de Referencia de su Computador, o al final de este libro.

USO DE LA TECLA CONTROL [CTRL] 2a. parte

Anteriormente hemos visto que, el uso de la tecla [CTRL] permite la escritura de una serie de caracteres gráficos, y la posibilidad de desplazar el cursor en las cuatro direcciones, con el fin de cambiar el texto o corregirlo.

Ahora veremos algo más.

a) Escriba: 10 PRINT "LO QUEL VIENTO SE LLEVO"

Observe que escribió QUEL en lugar de QUE EL. ¿Cómo corregir? Una solución obvia, sería reescribir la línea 10. Pero existe otra.

Lleve el cursor a la línea 10 y ubíquelo sobre la letra L de QUEL. Mantenga oprimida la tecla [CTRL] mientras presiona [INSERT]. El texto, partiendo desde la L sobre la que está el cursor, se desplazará un lugar a la derecha, pero el cursor permanecerá en el mismo sitio. Entonces oprima [INSERT] una vez más, para dejar el espacio correspondiente entre QUE y EL. A continuación, avance el cursor un lugar a la derecha y escriba E. Luego [RETURN].

Si al presionar [INSERT] el texto se desplazó a la derecha más de lo necesario, presione [CTRL][DELETE] para retroceder el texto hasta el lugar adecuado y luego [RETURN].

El ejemplo anterior, le servirá para captar la mecánica de la operación. Repita el ejemplo, varíelo, arregle y desarregle textos, en fin, practique y finalmente podrá apreciar las ventajas de esta "edición de pantalla".

NOTA: Si Ud. no oprime [RETURN] después de hacer los cambios necesarios, sólo habrá modificado la pantalla y no la línea de programa en la memoria del computador.

b) Escriba: 10 PRINT "HOLA":GOTO 10
RUN

Cómo detener la escritura en pantalla?

Presione [CTRL][!/1]. Se detuvo.

Presione [CTRL][!/1]. Volvió a correr.

El uso de esta combinación de teclas es útil para detener, y luego reanudar, la ejecución o el listado de un programa, en forma temporal.

c) Duplicar líneas.

A veces es necesario repetir una instrucción determinada, en una o más partes del programa, o bien, puede suceder que su programa se haya llenado con más instrucciones, Ud. ya no tenga espacio para intercalar instrucciones y deba reenumerar. En tal caso, puede recurrir a lo siguiente:

Supongamos que la instrucción
 30 FOR R=1 TO 50:NEXT R (escribala),
 también la necesita en las líneas 60 y 90.

- Lleve el cursor al 3 de la línea 30. Levante su dedo de [CTRL] y oprima el 6, [RETURN].
- Lleve el cursor al 6 de la línea 60. Levante su dedo de [CTRL] y oprima el 9, [RETURN].

Al escribir LIST debería obtener en la pantalla lo siguiente:

```
30 FOR R=1 TO 50:NEXT R
60 FOR R=1 TO 50:NEXT R
90 FOR R=1 TO 50:NEXT R
```

En caso de que sea necesario reenumerar una parte del programa, proceda de igual forma, teniendo cuidado de borrar previamente de la memoria la instrucción a reenumerar.

Ej.: Si desea cambiarle el número a la línea 15, proceda de la siguiente manera:

- PRIMERO: escriba LIST, [RETURN].
- SEGUNDO: escriba 15, [RETURN]. Con esto ha borrado la línea 15 de la memoria del computador.
- TERCERO: Lleve el cursor hasta la línea 15 del listado que tiene en pantalla. Escriba, sobre el 15, el número que Ud. desea asignarle a la instrucción, y [RETURN].

En seguida, haga el listado del programa nuevamente y verá que: ya no hay una línea 15 en su programa y, la línea que tenía el número 15, ahora tiene el nuevo número.

VARIABLES

El computador acepta dos clases de variables, las variables numéricas (números) y las alfanuméricas o variables alfabéticas (letras).

En general, una VARIABLE es una magnitud o entidad sujeta a cambios o variaciones, es decir que puede tomar valores distintos a lo largo del tiempo.

De modo más simple podemos decir que una VARIABLE es cualquier cosa que puede cambiar o variar.

Algunas variables familiares para el común de las personas son:

- el color de una pantalla de TV
- la edad
- el tiempo
- la tasa de inflación
- el precio del oro.

a) Variables numéricas.

Son representadas por una letra o un conjunto de letras y números. A este tipo de variables sólo es posible asignarle un valor numérico.

Ej.: A=25; AB=16; A1=2; FIN=33; etc.

b) Variables alfanuméricas.

Están representadas por una letra o conjunto de letras, números o signos, seguidos por el símbolo \$. A este tipo de variables es posible asignarles valores numéricos o valores alfanuméricos, a condición de que vayan entre comillas.

Ej.: A\$= "RAM LTDA. INGENIERIA Y COMPUTACION."
 B2\$= "EL VALOR DEL DOLAR ES:"
 JK\$= "CIENTO DOS CONEJOS EN UN CAJON.
 CUANTAS PATAS Y OREJAS SON?"

Las variables alfanuméricas se nombran:

A\$ = A string o A pesos o A cadena.

AB\$ = AB string o AB pesos o AB cadena.

Al comienzo el computador le asigna el valor 0 (cero) a todas las variables, hasta que Ud. las defina dándoles un valor.

Resumiendo, existen dos tipos de variables:

Variables numéricas

A = 17.15
PRECIO = 253.32
EDAD = 33

Variables alfanuméricas

NOMBRE\$ = "JUAN PEREZ"
COMPUTADOR\$ = "ATARI"
AUTO\$ = "PEUGEOT"

Respecto de las variables alfanuméricas, hay que hacer un par de observaciones importantes.

a) En algunos computadores, entre los cuales se cuenta ATARI, es necesario DIMENSIONAR las variables alfanuméricas. DIMENSIONAR significa decirle al computador que deberá reservar un determinado espacio en la memoria, para guardar los caracteres que constituirán la variable. Esto se realiza mediante la instrucción DIM, indicando, además, el número máximo de caracteres que contendrá la variable alfanumérica o "string".

Por ejemplo: 10 DIM AUTO\$ (7)

b) La segunda cosa importante que cabe destacar, y que no debe olvidarse, es que el contenido del string DEBE ir siempre ENTRE COMILLAS.

Ejemplo:

```
10 DIM AUTO$ (7)
20 LET AUTO$ = "PEUGEOT"
30 PRINT AUTO$
```

Ejemplo:

```
10 DIM GRAFICO$ (10)
20 GRAFICO$ = "[CTRL]T [CTRL]T [CTRL]T [CTRL]T"
30 PRINT GRAFICO$
```

Aprovechando los gráficos del teclado, se pueden dibujar cosas muy entretenidas.

Ensaye el siguiente programa:

(coloque los espacios adecuados) U = [CTRL]U

```
10 PRINT "UUUU UU UU UU U"
20 PRINT "U U U U U U U"
30 PRINT "U U U U U U U"
40 PRINT "UUUU U UU UU U"
50 PRINT "U U U U U U U"
60 PRINT "[CTRL]T": GOTO 60
```


INSTRUCCION LET, DIM

Estas instrucciones están íntimamente relacionadas con las variables.

LET.

Esta instrucción le permite asignar un cierto valor a una variable determinada.

Ejemplo:

```
10 LET K=25
```

```
10 LET K$= "KILO"
```

La especificación LET, puede obviarse en el lenguaje ATARI BASIC. Sin embargo, en el BASIC de otros computadores, debe indicarse explícitamente.

DIM.

Permite reservar memoria para poder manejar los strings.

Ej.: 10 DIM NOMBRE\$ (6)

Esto significa que estamos reservando 6 bytes en la memoria, para poder alojar el valor de la variable NOMBRE\$. El valor de NOMBRE\$ puede tener como máximo 6 caracteres.

```
Por ejemplo: NOMBRE$= "PEDRO"
              NOMBRE$= "ISABEL"
              NOMBRE$= "MAGDALENA"
              PRINT NOMBRE$
```

```
Tiene 5 caracteres.
Tiene 6 caracteres.
Tiene 9 caracteres.
el computador lo acepta,
pero sólo guarda los
primeros 6 caracteres
(MAGDAL).
```


INSTRUCCION INPUT (INGRESE)

INPUT es una instrucción que permite al computador recibir instrucciones directas desde el teclado. Cuando el computador ejecuta una instrucción INPUT, espera una señal proveniente del teclado. Mientras no reciba ese "input" que espera, no ocurrirá nada más.

Ejemplo:

```
10 REM USO DE INSTRUCCION INPUT
20 DIM N$(10)
30 PRINT "CUAL ES TU NOMBRE?": INPUT N$
40 PRINT "HOLA ";N$, " COMO ESTAS?"

10 PRINT "ESCRIBE TU EDAD"
20 INPUT E
25 POKE 756,204: REM ESTO PERMITE EL ACCESO AL SET
  DE CARACTERES INTERNACIONALES
30 PRINT "TIENES ";E;"A[CTRL][B]OS"
```

RECUERDE: si usa variables alfanuméricas (STRINGS), antes debe dimensionarlas.

Su computador ATARI es uno de los computadores que escribe realmente castellano, con acentos, eñe y símbolos de pregunta correctos. Entre los minicomputadores es el único y no solo eso, además escribe en italiano, francés, portugués, griego, etc.

INSTRUCCION GOTO (VAYA A)

GOTO es una instrucción que se explica por sí misma. Significa GO= VAYA, TO= A, o sea: VAYA A. Al ejecutarla, el computador va a la línea cuyo número aparece después de la instrucción, y continúa ejecutando el programa desde esa línea en adelante.

GOTO es una instrucción incondicional. Esto quiere decir que, al encontrarla, el computador va, irrevocablemente, a ejecutar la línea a la que se le envió.

Ejemplo:

```
10 A=5
20 GOTO 90
30 B=10
40 C=20
50 PRINT B*C
60 END
90 PRINT A
100 GOTO 30
```

CONTADORES O ACUMULADORES.

Una aplicación muy común de la instrucción GOTO, es en los llamados CONTADORES.

Un contador, o acumulador, es una estructura de programación que cambia el valor de una variable, cada vez que el computador pasa por ahí, durante la ejecución de un programa.

Cuando está asociado a un bucle, cuenta cuantas veces se ejecuta el bucle en cuestión.

Ejemplo:

```
10 X=1
20 PRINT X
30 X= X+1          (no es una ecuación algebraica)
40 GOTO 20
```

La línea 10 inicializa la variable X, le da un valor de partida. En caso de no hacerlo, el computador asume que el valor inicial es cero.

La línea 30 indica que debe sumar 1 al valor actual de X. NO-TE QUE ESTO NO ES UNA ECUACION ALGEBRAICA!

NOTA: Bucle es una estructura de programación constituida por una instrucción o conjunto de instrucciones que se ejecuta muchas veces en forma sucesiva.

Ejemplo:

```

10 REM CHEQUERA
20 REM NUM ES EL NUMERO DEL CHEQUE
30 REM CANT ES EL VALOR DEL CHEQUE
40 REM TOT ES EL TOTAL DE DINERO ACUMULADO
50 NUM=1: CANT=0: TOT=0
60 PRINT "INGRESE CANTIDAD DEL CHEQUE #";NUM:
  INPUT CANT
70 NUM= NUM+1: TOT= TOT+CANT
80 PRINT: PRINT "UD. HA GIRADO $";TOT
90 PRINT: GOTO 60

```

Hay algo más que debemos hacer notar. En ambos ejemplos, especialmente en el primero, el bucle no tiene fin y, rápidamente, se alcanzará la saturación de la capacidad numérica del computador. Si lo desea, puede verificar que la capacidad de un computador ATARI alcanza hasta $1 \cdot 10^9$.

Naturalmente, estos ejemplos son solo eso, ejemplos. Sin embargo, en un programa real, Ud. debe tener en cuenta esta posibilidad de desastre al trabajar con bucles, y tomar las precauciones del caso.

INSTRUCCION ON/GOTO (DE ACUERDO CON/ VAYA A)

Esta es una instrucción muy útil para asociar expresiones alfanuméricas (instrucciones, mensajes, fórmulas, etc.) con el valor de una variable.

El formato de esta instrucción es:

ON X GOTO 30, 80, 120, (etc.)

X= cualquier variable numérica

30= número de línea a ejecutar cuando X=1

80= número de línea a ejecutar cuando X=2

120= número de línea a ejecutar cuando X=3

Esta instrucción permite la inclusión de MENUS en los programas.

Ejemplo:

```

10 POS. 10,5: PRINT "1. - CHILE"
20 POS. 10,7: PRINT "2. - SANTIAGO"
30 POS. 10,9: PRINT "3. - PROVIDENCIA"
40 POS. 10,11: PRINT "4. - AV. R. LYON"
45 PRINT "ESCOGE UNA";: INPUT R
50 ON R GOTO 60, 70, 80, 90
60 PRINT "ES UN PAIS DE SUDAMERICA": GOTO 100
70 PRINT "ES LA CAPITAL DE CHILE": GOTO 100
80 PRINT "ES UNA COMUNA DE SANTIAGO": GOTO 100
90 PRINT "ES UNA CALLE DE PROVIDENCIA": GOTO 100
100 END

```


INSTRUCCION FOR/NEXT (PARA/PROXIMO)

Esta instrucción es un ejemplo típico de lo que se denomina LOOP o bucle. Un bucle es una estructura de programación, constituida por una instrucción o conjunto de instrucciones que se ejecuta muchas veces, en forma sucesiva.

La instrucción FOR/NEXT permite repetir una secuencia las veces que uno quiera, con gran facilidad y sin mucho gasto de memoria.

Escriba:

```
10 FOR J=1 TO 25
20 PRINT "J=";J
30 NEXT J
40 PRINT "FIN"
```

y corra el programa. [RUN]

Como puede verse, el computador cuenta de 1 a 25 y lo escribe en la pantalla.

Cómo ha sucedido esto? El computador ha empezado por ejecutar la línea 10, que dice FOR J=1 TO 25. Empieza haciendo J=1, luego la línea 20 le dice PRINT "J=";J, por lo que escribe J=1. En seguida, la línea 30 le indica NEXT (= próximo) J. Como hasta ese momento J=1, ahora hace J=2 y va de nuevo a la línea 10. Como J=2 está en el rango especificado en 10, continúa con 20, y así sucesivamente, hasta que escribe J=25, obedeciendo a la línea 20. Inmediatamente después, en la línea 30, le dicen hacer NEXT J, o sea que haga J=26, pero esta vez J=26 cae fuera del rango especificado en 10, y ya no repite la instrucción 20. Entonces, sale del LOOP y ejecuta la línea 40.

Así vemos que el computador repite, un número determinado de veces, las instrucciones que están ubicadas entre el FOR y el NEXT.

Otro ejemplo:

```
10 FOR X=1 TO 5
20 Y= 2X+1
30 PRINT X, Y
40 NEXT X
```


Ahora, intente este pequeño programa:

```
10 REM APUESTO QUE NO LO VE
20 GR.19: COLOR 1: PLOT 19,11
30 END
```

¿Qué sucedió? No ve nada. El computador es demasiado rápido.

Escriba:

```
30 FOR R=1 TO 500: NEXT R
40 END
```

Corra el programa.

Ahora si se ve algo, durante algunos segundos.

La línea 30 ha introducido un "LOOP VACIO", es decir un retardo. Al llegar al 30 el computador empieza a contar desde 1 a 500, sin hacer nada más y, hasta que no termine, no sigue a la línea 40. Este retardo que, en buenas cuentas, dura lo que el computador demora en contar de 1 a 500, es lo que permite visualizar en pantalla el significado de las instrucciones de la línea 20. Estas instrucciones son típicas de un programa de gráficos que veremos más adelante.

BUCLES INCLUIDOS.

Hemos señalado que FOR/NEXT es una estructura de programación. También se le llama BUCLE DE PROGRAMA, dado que, durante la ejecución del programa, el computador da vueltas entre el FOR, el NEXT y nuevamente el FOR. Esta estructura es muy común. La mayoría de los programas escritos en BASIC cuentan con dos o más de estos bucles. En muchos casos, muchísimos, es posible encontrar bucles dentro de otros bucles más grandes. En estos casos se habla de BUCLES INCLUIDOS (NESTED LOOPS). Es muy sencillo trabajar con estos bucles incluidos, sólo hay que tener mucho cuidado de no terminar un bucle externo, antes de haber terminado un bucle interno.

Antes de ejemplificar la situación anterior, conviene aclarar algo que se nos estaba quedando atrás.

```
10 FOR K=1 TO 100
```

En esta instrucción se está dando por sobreentendido que la variación de K, entre 1 y 100, es de 1 en 1 (1, 2, 3, etc.). Sin embargo, también es posible hacer que K varíe de 5 en 5, de 10 en 10 o de 20 en 20. Para ello:

```
10 FOR K=1 TO 100 STEP 10
```


De este modo, K variará de 10 en 10, es decir: la 1a. vez será K=1, la 2a. K=11, la 3a. K=21, y así hasta K=91. El siguiente K, K=101, queda fuera del rango y no se ejecuta.

Para que la cuenta vaya 10, 20, 30, ..., 100..
 10 FOR K=10 TO 100 STEP 10

También es posible hacer variar K, disminuyendo su valor.
 10 FOR K=100 TO 10 STEP -10

Ejemplo de programa con bucles FOR/NEXT.

```
10 FOR Y=1 TO 3
20 FOR K=255 TO 0 STEP -5
30 SOUND 0, K, 10, 10
40 PRINT K
50 FOR Z=1 TO 30: NEXT Z: REM RETARDO
60 NEXT K
70 NEXT Y
```

Ejercicios: a) escriba su nombre 10 veces.
 b) escriba la tabla del 9 (de multiplicar)

INSTRUCCION IF/THEN (SI ACASO/ ENTONCES)

IF/THEN es la instrucción que le permite al computador tomar DECISIONES. Es una instrucción condicional, que significa SI/ENTONCES.

En términos de programación, se escribe de la forma:

IF (si)... "expresión"... THEN (entonces)... "instrucción"...
y debe entenderse como:

SI la "expresión" es VERDADERA.
ENTONCES ejecute la "instrucción".

Ejemplos:

```
10 IF A=B+5 THEN PRINT UR$
7 IF K=0 THEN GOTO 70
80 IF NOMBRE$= "HERMES" THEN A=A+1
```

Cada uno de estos ejemplos podría formar parte de un programa diferente. Cada uno establece una condición que, si se cumple, obliga al computador a ejecutar la instrucción que sigue al THEN. Pero la ejecutará si y solo si la condición se cumple. De otro modo el computador continuará con la ejecución de la línea siguiente. Tenga presente que, si en la misma línea física hay otra línea de programa a continuación del IF/THEN, ésta quedará sin ejecutarse si la condición no se cumple.

SIMBOLOS DE COMPARACION

IF A > B	THEN	significa	SI A es mayor que B	ENTONCES...
IF A < B	THEN	significa	SI A es menor que B	ENTONCES...
IF A = B	THEN	significa	SI A es igual a B	ENTONCES...
IF A >= B	THEN	significa	SI A es mayor o igual que B	ENTONCES...
IF A <= B	THEN	significa	SI A es menor o igual que B	ENTONCES...
IF A <> B	THEN	significa	SI A es distinto de B	ENTONCES...

Ejemplo:

```
10 IF A>B THEN PRINT "A ES MAYOR"
20 PRINT "ENTONCES A<=B"
```

En la línea 10 el computador compara A con B. Si A es mayor que B, entonces ejecuta la instrucción PRINT que sigue a THEN. Si A no es mayor que B (o sea, si es menor o igual), continúa la ejecución del programa en la línea 20.

Ejemplos:

```
10 REM ADIVINANZA
20 DIM A$(6)
30 PRINT "DE QUE COLOR ES EL CABALLO BLANCO
DE NAPOLEON?";:INPUT A$
40 IF A$= "BLANCO" THEN GOTO 100
50 PRINT "NO, TE EQUIVOCASTE. TRATA DE NUEVO."
60 GOTO 30
100 PRINT "BRAVO! ACERTASTE.": END
```

```
5 REM SIMULAR LANZAMIENTO DE DADOS
```

```
10 DIM A$(2)
```

```
15 PRINT "[ESC]-[CTRL] [CLEAR]"
```

(en la pantalla aparecerá una flecha curvada hacia arriba y a la izquierda)

```
20 X= INT (RND(0)*6)+1
```

```
30 Y= INT (RND(0)*6)+1
```

```
40 R= X+Y
```

```
50 POS. 12,6: PRINT X
```

```
60 POS. 25,6: PRINT Y
```

```
70 POS. 15,11: PRINT "HA OBTENIDO";R
```

```
80 POS. 9,14: PRINT "QUIERE JUGAR OTRA VEZ?";: INPUT A$
```

```
90 IF A$= "SI" THEN 20
```

```
100 END
```

NOTA: la línea 15 le dice al computador que le limpie la pantalla.

Juega muy rápido? Bueno, entonces arréglo.

Ayuda: Introduzca bucles vacíos en los lugares adecuados. Ensaye, experimente!

NOTA: Habrá observado que en la línea 90, después de THEN, hemos escrito sólo 20 en lugar de GOTO 20.

El ATARI BASIC permite dos formas de IF/THEN:

```
90 IF A$= "SI" THEN GOTO 20
```

```
90 IF A$= "SI" THEN 20
```

También se pueden probar varias condiciones a la vez con el uso de IF/THEN, combinado con "AND" u "OR", o varias pruebas consecutivas de IF/THEN.

Ejemplo:

```
10 A= 20: B= 30
```

```
20 IF A=20 AND B>20 THEN PRINT "A Y B SON MAYORES QUE 20"
```


Para probar ambas condiciones experimente con los valores de A y B en la línea #10, o escriba su propio programa para entrar los valores de A y B.

Haga un programa para probar "OR".

Ejemplo:

```
10 A= 5: B=4
20 IF A+B=9 THEN IF A-B=1 THEN PRINT "CORRECTO"
30 END
```

Experimente.

Finalmente, IF/THEN puede usarse para comprobar la existencia (y el uso) de una variable en el programa. Por ejemplo:

```
10 A=1
20 IF A THEN PRINT "AFIRMATIVO"
30 END
100 REM IMPRIME AFIRMATIVO SIEMPRE QUE A<> 0.
```

```
10 A=0
20 IF NOT A THEN PRINT "NEGATIVO"
30 END
100 REM IMPRIME NEGATIVO SIEMPRE QUE A=0.
```


INSTRUCCION GOSUB/RETURN (VAYA A SUBROUTINA/ REGRESE)

La instrucción GOSUB es muy parecida a la instrucción GOTO. También salta a una línea de programa especificada en la instrucción. Sin embargo, existe una diferencia importante:

GOSUB va a la línea de programa especificada y, además, toma nota de donde debe volver. Volverá a la instrucción que sigue a GOSUB. Si GOSUB es la última instrucción en una línea de programa, volverá a la primera instrucción de la línea siguiente.

SUBROUTINA es una sección definida del programa, a la cual se acude con el fin de ejecutar una función u operación determinada y luego retornar al programa principal. La subrutina puede considerarse como un "subprograma", del programa principal, que generalmente se repite varias veces.

La instrucción que permite volver al programa principal es RETURN.

Toda subrutina debe terminar con la instrucción RETURN. Al ejecutar RETURN, la acción del computador vuelve a la instrucción que sigue al GOSUB.

Ejemplo:

```
10 REM COMO FUNCIONA UNA SUBROUTINA.
20 REM PROGRAMA PRINCIPAL
30 PRINT "NECESITO MULTIPLICAR X*Y"
40 INPUT X, Y
50 GOSUB 200
60 PRINT "VOLVI DE LA SUBROUTINA"
70 PRINT "X*Y= ";T
80 END
200 REM SUBROUTINA DE MULTIPLICACION
210 R=X: S=Y
220 T= R*S
230 RETURN
```

Una persona que piensa programar en forma intensa, debería hacerse un archivo de subrutinas para las tareas repetitivas de cada programa. Por ejemplo, limpiar la pantalla, desactivar algunas funciones, manejo de errores, archivos en cassette o diskette, carátulas de programas, música típica de sus programas, etc.

INSTRUCCION ON/GOSUB (DE ACUERDO CON/ VAYA A SUBROUTINA)

Así como, a partir de GOTO, tenemos ON/GOTO, a partir de GOSUB obtenemos ON/GOSUB. Tanto la lógica, como el formato, son similares a los de ON/GOTO.

ON X GOSUB 1000, 700, 1500, etc.

X= variable numérica.

1000= línea en que comienza la subrutina a usar si X=1.

700= línea en que comienza la subrutina a usar si x=2.

1500= línea en que comienza la subrutina a usar si x=3.

Naturalmente, cada subrutina debe terminar con una instrucción RETURN. Después de ejecutar RETURN, el computador ejecuta la instrucción que sigue a GOSUB.

Ejemplo:

```
10 REM ON/GOSUB
20 PRINT:PRINT "PREGUNTA #1"
30 PRINT "CUAL ES LA RAIZ CUADRADA DE 100?"
40 INPUT RC:IF RC=10 THEN GOTO 50
45 PRINT "INTENTE DE NUEVO":GOTO 30
50 HAL=INT(RND(0)*3)+1:ON HAL GOSUB 600,700,800:GOTO 70
60 REM DOS PREGUNTAS EN EL EJEMPLO
70 PRINT:PRINT "PREGUNTA #2"
80 PRINT "CUANTOS PARES SON TRES MOSCAS?"
90 INPUT CP: IF CP=3/2 THEN 100
95 PRINT "INTENTE DE NUEVO":GOTO 80
100 HAL=INT(RND(0)*3)+1:ON HAL GOSUB 600,700,800
120 PRINT:PRINT "ADIOS": END
600 PRINT "BUEN TRABAJO!":RETURN
700 PRINT "OTRO EINSTEIN!":RETURN
800 PRINT "CORRECTO!":RETURN
```


INSTRUCCION READ/DATA (LEA/ DATOS)

Cuando hablamos de las variables, vimos una forma de asignarles valores. Esto se hace mediante la instrucción LET o, como en el caso del ATARI BASIC, sin ella.

Recordemos:

```

      10 LET X=25
o     10 X=25

```

Cuando se trata de asignar valores (o data) a un número de variables, mayor que 4 o 5, puede resultar más conveniente utilizar las instrucciones READ/DATA.

Sin embargo, comencemos con un ejemplo sencillo, explicando como el computador lee y asigna valores a una variable.

Escriba:

```

10 DATA 5,10,15,20,25
20 READ A
30 PRINT "A=";A
40 GOTO 20

```

En la primera pasada, el computador toma A=5, lo escribe en pantalla y vuelve a la línea 20. En la segunda pasada, asume A=10, lo muestra en la pantalla y regresa otra vez a la línea 20. De esta forma, se le asignan a A, los cinco valores que aparecen en la instrucción DATA. El computador ha leído la información y se la ha asignado a una variable. Es como si hubiera usado un PUNTERO para ir indicando qué valores le corresponden a A en cada caso.

Corra el programa. Se obtuvo lo esperado, verdad? Pero además, obtuvimos algo inesperado.

ERROR- 6 AT LINE 20

Es culpa del puntero. Después de la quinta pasada el computador volvió a la línea 20, pero ya no tiene más datos que leer, de modo que el "puntero" quedó apuntando al aire. Esto significa que el computador se quedó sin información, es decir, se le acabaron los valores que debía asignarle a la variable, y nadie le dijo nada.

La forma más usual de evitar esta situación, es asignarle a

la variable un valor que no tenga nada que ver, que sea absolutamente distinto de los valores que pueda tomar normalmente. A este valor se le ha llamado "FLAG", y se emplea en combinación con IF/THEN.

Escriba:

```
10 DATA 5,10,15,20,25,0.115
20 READ A
25 IF A=0.115 THEN END
30 PRINT "A=";A
40 GOTO 20
```

Ahora veamos el siguiente ejemplo:

```
10 DATA 5,10,15,20,25
20 READ A,B,C,D,E
30 PRINT "A=";A, "B=";B, "C=";C, "D=";D, "E=";E
```

Esta vez el computador asignó las variables respectivas en el orden en que aparecen las instrucciones correspondientes.

También se puede operar con variables alfanuméricas, teniendo cuidado de asignar el valor o data adecuados, a la variable que corresponda. Es decir:

- números a variables numéricas
- caracteres alfanuméricos a variables alfanuméricas

Ejemplo:

```
10 DIM A$(10), B$(10)
20 READ A, A$, B, B$, C
30 PRINT A, A$, B, B$, C
40 PRINT A$;" TIENE ";A;" AGNOS"
50 PRINT "EL ";B;" IRE AL ";B$
60 DATA 15, AURORA, 18, CORREO, -0.01
```

Como se ve, la línea de DATA puede ir al comienzo o al final del programa, como más le convenga.

INSTRUCCION RESTORE (REPONGA)

Este comando se encuentra asociado a READ/DATA. Después de ejecutar una instrucción READ, el "puntero" queda apuntando al último dato, valor o información, que se leyó. La instrucción RESTORE, lo devuelve al principio.

Agregue

70 RESTORE: GOTO 20

al último ejemplo, y observe que sucede. Ahora pruebe eliminando RESTORE.

RESTORE le permite dirigirse a una línea específica de datos (al primer dato de dicha línea).

Por ejemplo:

```
10 DATA 1,2,3,4,5
20 DATA 10,20,30,40,50
30 READ A
40 PRINT A
49 REM MUEVA EL PUNTERO A LA LINEA 20
50 RESTORE 20
60 READ A
70 PRINT A
```


ARRAYS (ARREGLOS)

Se le llama ARRAY a una forma sistemática de designar un gran número de variables.

Puede pensarse en ellos como una serie de "cajas" o ELEMENTOS en los cuales podrá guardar números.

Cada una de estas cajas o ELEMENTOS es una variable numérica y, como variable que es, Ud. puede asignarle valores, cambiárselos o efectuar cualquier operación matemática con ella.

Sin embargo, deben respetarse algunas reglas:

REGLA # 1. Cada ELEMENTO del ARRAY, ej. A(25), tiene un nombre compuesto de dos partes:

- Una letra o letras (A, AB, etc.), que constituyen el nombre del ARRAY.
- Un número, que va entre paréntesis y recibe el nombre de "INDICE" o "SUBSCRIPTO". Este indica el número de orden del elemento en el ARRAY.

Supongamos un conjunto de 5 elementos. A este conjunto lo llamaremos ARRAY y su nombre será A. Además, le asignaremos un valor a cada elemento.

Entonces tendremos:

```
A(0)=3
A(1)=-2.5
A(2)=10
A(3)=0.45
A(4)=53
```

REGLA # 2. Antes de utilizar un ARRAY debemos DIMENSIONARLO. Ya antes hemos DIMENSIONADO strings. Existe una gran diferencia entre DIMENSIONAR ARRAYS y DIMENSIONAR STRINGS.

Al dimensionar un string, estamos indicando el número máximo de caracteres que va a contener.

Al dimensionar un ARRAY, indicamos el número máximo de ELEMENTOS que tendrá, junto con indicar el nombre de dichos elementos. Además, en cada elemento se guardará información (data, números).

Así, el conjunto de elementos definido anteriormente, será

conocido como el ARRAY A(5) y, para dimensionarlo, se escribirá: DIM A(5).

Bien, y ahora, que podemos hacer con esto?

Un uso inmediato es la asignación de valores a un gran número de variables, que de otro modo se haría mediante asignación directa (A1=0, A2=1, A3=2, etc.) o mediante la instrucción READ/DATA.

Utilizando los ARRAYS, esto sería:

```
10 DIM A(50)
20 FOR K=0 TO 49
30 A(K)= K+1
40 NEXT K
```

Corra el programa y escriba, por ejemplo, PRINT A(25). El computador entregará 26. Esto significa que, en el elemento A(25) del ARRAY, el computador ha guardado el número 26.

Agregue la siguiente instrucción en la línea 30 (no olvide separarla con :).

```
PRINT "A(";K;") = ";A(K)
```

Así obtiene todos los elementos del ARRAY, y los valores que se asocian a ellos.

MATRIX (MATRICES)

Estas son una versión más poderosa y complicada de los arrays. Una matriz es una red en que definimos FILAS y COLUMNAS.

- FILAS, en el sentido horizontal, y
- COLUMNAS, en el sentido vertical.

Cada cruce entre fila y columna, contendrá un elemento de la matriz y, en ese elemento, podremos guardar un número.

$$M = \begin{matrix} & 1 & 2 & 3 \\ & 4 & 5 & 6 \\ & 7 & 8 & 9 \end{matrix}$$

es una matriz de tres filas y tres columnas, que contiene nueve elementos.

Cada elemento está definido por una expresión de la forma $M(i,j)$, donde $i = 1, 2$ o 3 , y $j = 1, 2$ o 3 .

Así pues:	$M(1,1) = 1$	$M(1,2) = 2$	$M(1,3) = 3$
	$M(2,1) = 4$	$M(2,2) = 5$	$M(2,3) = 6$
	$M(3,1) = 7$	$M(3,2) = 8$	$M(3,3) = 9$

Claramente se advierte que en la expresión $M(i,j)$, i representa las filas y j las columnas.

La matriz no tiene por qué tener el mismo número de filas que de columnas.

Cualquier problema que pueda ser planteado en forma de líneas y de columnas, puede ser tratado o programado mediante matrices.

Por ejemplo, supongamos una clase con 5 estudiantes. Cada estudiante tiene que pasar 3 pruebas y cada prueba tendrá una nota.

Estudiantes:	del 1 al 5	(filas)	
Pruebas	: del 1 al 3	(columnas)	
Notas	: el valor asignado a cada elemento		
Matriz	: EP		

EP(1,1) -	EP(1,2) -	EP(1,3) -
EP(2,1) -	EP(2,2) -	EP(2,3) -
EP(3,1) -	EP(3,2) -	EP(3,3) -
EP(4,1) -	EP(4,2) -	EP(4,3) -
EP(5,1) -	EP(5,2) -	EP(5,3) -

Las notas deben ir en el lugar donde están los guiones. Estas notas son los elementos que se han asignado a los elementos de la matriz.

Para poner los valores de las notas usaremos la instrucción INPUT.

Aquí va el programa:

```

10 DIM EP(5,3)
20 FOR J=1 TO 3
30 FOR I=1 TO 5
40 PRINT "PONGALE NOTA AL ESTUDIANTE EP(";I;",";J;")
Y OPRIMA RETURN."
50 INPUT CALIFICACION
60 EP(I,J)= CALIFICACION
70 NEXT I
80 NEXT J
90 PRINT "[ESC]-[CTRL][CLEAR]"
100 FOR X=1 TO 3
110 FOR Y=1 TO 5
120 IF X=1 THEN POSITION 2,Y
130 IF X=2 THEN POSITION 15,Y
140 IF X=3 THEN POSITION 28,Y
150 PRINT "EP(";Y;",";X;")=";EP(X,Y)
160 NEXT Y
170 NEXT X

```


SEGUNDA PARTE

GRABADORA DE PROGRAMAS

La grabadora de programas o "cassettera", es un aparato que permite grabar programas, escritos en BASIC y que están alojados en la memoria del computador, en cassettes (cintas magnéticas). Posteriormente, los programas grabados en cassettes pueden ser cargados en la memoria del computador, mediante el mismo aparato.

Hay 3 comandos ATARI BASIC para grabar programas en cassettes: CSAVE, LIST y SAVE. Cada uno de ellos tiene su contrapartida: CLOAD, ENTER y LOAD, que le permiten cargar programas en la memoria de su computador.

GRABAR UN PROGRAMA.

CSAVE es un comando especial para grabar programas en cassettes, solamente. LIST y SAVE son comandos generales que hacen salir (sacar o enviar) un programa a cualquier dispositivo de salida, la grabadora es tan solo uno de ellos. Cualquiera de los siguientes comandos grabará un programa en la grabadora:

- a) CSAVE
- b) SAVE"C:"
- c) LIST"C:"

Observe que, en los casos b) y c), debe especificarse la grabadora de programas (dispositivo C:).

Los comandos CSAVE y SAVE"C:" graban SIEMPRE el programa entero contenido en memoria. El comando LIST"C:" puede grabar una parte o la totalidad del programa contenido en la memoria.

Tal como con otras formas del comando LIST, Ud. puede especificar la primera y la última línea que desea listar. El siguiente comando hará que las líneas de programa cuyos números están comprendidos entre 125 y 1250, sean pasadas a la grabadora.

LIST"C:",125,1250

Los tres comandos mencionados provocan la misma secuencia de hechos. Escriba cualquiera de ellos en la pantalla y presione RETURN.

El computador emitirá dos sonidos (pitazos). Esta señal indica que el computador ha comprendido el mensaje, y que Ud. debe preparar el cassette y la grabadora para efectuar la operación.

Con las teclas REWIND y FAST FORWARD, Ud. está en condiciones de poner la cinta en la posición a partir de la cual quiere empezar a grabar. Generalmente será desde el principio de la cinta. A continuación, presione en forma conjunta las teclas RECORD y PLAY. (Dos pitazos para grabar con dos teclas, RECORD y PLAY.)

NOTA: Presione suavemente las teclas. Si tiene dificultades para hacer enganchar la tecla RECORD, verifique si el cassette está protegido. Cambie de cinta o, si desea anular lo escrito y grabar encima, tape el orificio de protección.

El computador necesita saber cuando Ud. ha terminado de preparar la grabadora. Hágaselo saber oprimiendo [RETURN].

La cinta empieza a moverse. Si Ud. sube el volumen del televisor, escuchará como se desarrolla la grabación. Primero oirá un sonido agudo permanente, largo. En seguida, vienen uno o más estallidos de sonido, lo que significa que se están grabando los bloques de programa en el cassette. Mientras más largo sea el programa, más bloques se grabarán y se oirán más estallidos de sonido. Estos dejan de escucharse cuando se completa la grabación. La cinta se detiene y Ud. ya puede oprimir la tecla STOP. En la pantalla puede leerse READY.

Al proceder así con varios programas, no olvide tomar nota de las indicaciones del contador al principio y al final de la grabación, y dejar por lo menos 10 vueltas libres entre cada programa.

CARGAR UN PROGRAMA.

El comando LOAD permite leer y cargar en memoria un programa, que ha sido previamente grabado en cassette mediante el comando CSAVE. ENTER y LOAD son comandos generales que permiten cargar programas en memoria, desde cualquier periférico de entrada.

ENTER"C" permite cargar programas grabados con LIST"C:".
 LOAD"C:" permite cargar programas grabados con SAVE"C:".

Los comandos LOAD y CLOAD borran el programa que se halla en memoria, antes de cargar el nuevo. El comando ENTER"C:" no borra el programa en memoria, sino que lo mezcla con el nuevo. Si hay líneas de programa que tengan el mismo número, prevalecerán las del programa nuevo (el que se está cargando a la memoria). Para evitar esta mezcla, Ud. debe escribir NEW antes de cargar un programa con ENTER"C:".

Tal como en el caso de los comandos CSAVE, SAVE"C:" y LIST"C:" para grabar, el uso de los comandos para leer programas (CLOAD, LOAD"C:" y ENTER"C:") desencadenará la misma secuencia de acontecimientos.

Al escribir en la pantalla cualquiera de estos comandos y oprimir [RETURN], Ud. escuchará un solo pitazo (deberá apretarse una sola tecla).

Prepare la grabadora en su modo de operación para cargar programas, es decir use las teclas REWIND y FAST FORWARD para poner el contador en el número a partir del cual fue grabado el programa que Ud. quiere cargar ahora. Una vez hecho esto oprima la tecla PLAY.

Dígale al computador que ya terminó de preparar la grabadora. Oprima [RETURN].

La cinta empezará a moverse y Ud. podrá escuchar como se efectúa la carga del programa. Primero habrá un período de silencio de, aproximadamente, 20 segundos y, en seguida, uno o más estallidos de sonido. Cuando terminen, la cinta se detendrá y Ud. podrá oprimir la tecla STOP.

Ya tiene el programa cargado en la memoria (en la pantalla se lee READY). Ahora úselo.

CONTADOR DE VUELTAS.

El contador de vueltas, o simplemente el contador, está allí para ayudarle a encontrar el lugar donde Ud. tiene grabado tal o cual programa, siempre que Ud. haya procedido en forma sistemática y ordenada al grabar los programas.

Empiece siempre los cassettes desde el principio, poniendo el contador en cero.

Al terminar de grabar, anote el número que indica el contador.

Si necesita grabar otro programa, haga avanzar la cinta unas 10 vueltas, antes de empezar a grabar el nuevo programa. Nuevamente, anote el número que le indicará el contador.

HAGA LOS SIGUIENTES ENSAYOS:

1 Escriba:

```
10 PRINT "PROGRAMA A":PRINT
20 PRINT "ENSAYO DE INSTRUCCIONES PARA GRABAR PROGRAMAS EN
CASSETTE Y CARGARLOS DE VUELTA"
30 REM ENSAYO CON CSAVE/CLOAD
CSAVE
```

2 Escriba NEW. Luego CLOAD. Después que termine de cargar el programa, escriba LIST.

3 Escriba:

```
10 PRINT "PROGRAMA B":PRINT
30 REM ENSAYO CON SAVE"C: "/LOAD"C: "
SAVE"C: "
```

4 Escriba NEW. Luego LOAD"C: ". Después que termine de cargar el programa, escriba LIST.

5 Escriba:

```
10 PRINT "PROGRAMA C":PRINT
30 REM ENSAYO CON LIST"C: "/ENTER"C: "
50 PRINT "ESTA LINEA NO DEBE APARECER AL CARGAR EL PROGRAMA
DE VUELTA."
LIST"C: ",10,40
```

6 Escriba NEW. Luego ENTER"C: ". Terminada la carga, escriba LIST y verifique que la línea 50 no aparece en el listado. (No fue grabada en la cinta).

Si, al terminar el proceso de grabación, aparece un aviso de error, esto puede deberse a varias causas.

ERROR 138. Su grabadora está desconectada de la red eléctrica.

ERROR 130. Su grabadora no está conectada al computador.

ERROR 143. Errores debidos a ruidos en la línea de grabación. Restos de programas antiguos en la cinta o problemas con su grabadora. Pruebe de nuevo, varias veces y, si no resulta, haga revisar la grabadora. Asegúrese de que puso la cinta en el lugar adecuado (o unas tres vueltas antes). Hay veces en que la cinta se enrrolla más apretada que en otras, y entonces los contadores no reflejan la posición correcta.

ERROR 147. Se le acabó la memoria! Divida el programa o aumente su memoria.

NOTA 1: ATARI también cuenta con unidades de diskettes de alta capacidad de almacenamiento y gran velocidad de lectura, para manejo de archivos complicados.

NOTA 3: Para la sección de gráficos, que viene a continuación, Ud. debe estar bien familiarizado con la forma de programar su computador ATARI, es por eso que los ejemplos se muestran con abreviaciones y utilizando instrucciones múltiples por línea. Esta técnica le permite ahorrar memoria (bytes). Si le es difícil entender los programas, reemplace las abreviaciones con el nombre completo de la instrucción y, en cada signo :, divida la línea asignándole un nuevo número de línea, pero conservando el orden establecido en el ejemplo.

INTRODUCCION A GRAFICOS

En este capítulo nos referiremos a las capacidades de gráficos del computador ATARI.

Cualquiera sea la aplicación a que Ud. destine su computador ATARI, las potencialidades gráficas le servirán para realzar cualquier programa que haga.

El computador ATARI tiene 16 modos gráficos, numerados del 0 al 15. Estos modos se activan mediante la instrucción GRAPHICS, seguida del número correspondiente. Ej.: GRAPHICS 8.

De los 16 modos gráficos, 5 son de texto, es decir admiten la posibilidad de presentar o exhibir palabras en la pantalla, con distintos tipos de letras y colores.

Los modos de gráfico puro tienen distintas resoluciones, siendo menor la de GR. 3, también presentan ventana de texto a voluntad.

En caso de que no desee ventana de texto, sume 16 al número del modo gráfico en que trabajará, y desaparecerá la ventana de texto, extendiendo la pantalla.

Por ejemplo, si Ud. trabaja con GR. 5 y no precisa la ventana de texto, entonces trabaje con GR. 21 (o GR. 5+16).

OPCIONES DEL COMANDO "GRAPHICS"

Número Modo Gráfico	Suprimir Ventanilla de Texto	Suprimir Limpieza Pantalla	Suprimir Ambos
0	-	32	-
1	17	33	49
2	18	34	50
3	19	35	51
4	20	36	52
5	21	37	53
6	22	38	54
7	23	39	55
8	24	40	56
9	-	41	-
10	-	42	-
11	-	43	-
12*	28	44	60

13*	29	45	61
14*	30	46	62
15*	31	47	63

* Estos modos sólo los tiene la línea ATARI XL.

Las instrucciones que se emplean para trabajar con gráficos son:

- | | |
|-------------------|-------------|
| - SETCOLOR R,C,L | (SE. R,C,L) |
| - COLOR K (Kolor) | (C. K.) |
| - PLOT X, Y | (PL. X, Y) |
| - DRAWTO X, Y | (DR. X, Y) |

INSTRUCCION SETCOLOR. (JUEGO o GRUPO DE COLORES)

REGISTROS DE COLOR. En cualquier modo gráfico Ud. puede escoger uno o más colores para:

- el texto o gráficos (foreground o frente)
- el fondo (background)
- el borde, que enmarca al fondo.

El computador ATARI tiene REGISTROS DE COLOR, que son ubicaciones en la memoria, cuyo contenido controla el color del frente, fondo y borde.

La instrucción SETCOLOR permite seleccionar los registros y colores definidos para cada modo gráfico.

El formato de SETCOLOR es el siguiente:
SETCOLOR R,C,L

R = selecciona el registro de color que Ud. desea cambiar.

C = selecciona el color al que quiere cambiarlo.

L = selecciona el nivel de luminosidad de ese color.

R indica el número del registro de color. Los registros posibles son 5, y están numerados de 0 a 4.

C es el número que representa al color. Los colores están numerados del 0 al 15.

L determina la luminosidad y puede ir desde 0 (el más oscuro) hasta 14 (el más brillante o luminoso). Sólo los números pares tienen significado en este caso. El número puede ser mayor que 14, pero en este caso el computador dividirá por 16 y usará el resto para la luminosidad.

NUMEROS DE LOS COLORES USADOS CON SETCOLOR

Número	Color	Número	Color
0	Gris	8	Azul claro
1	Oro	9	Verde-Azul
2	Anaranjado	10	Agua
3	Rojo	11	Azul-Verde
4	Rosado	12	Verde
5	Violeta	13	Verde-Amarillo
6	Púrpura-Azul	14	Verde-Anaranjado
7	Azul	15	Anaranjado

VALORES PREASIGNADOS DE LOS REGISTROS DE COLOR

Número Registro	Número Color	Valor Lumnos.	Color Pantalla
Ø	2	8	Anaranjado
1	12	1Ø	Agua
2	9	4	Azul
3	4	6	Rojo Claro
4	Ø	Ø	Negro

INSTRUCCION COLOR (COLOR)

El formato de COLOR es el siguiente:

COLOR K

Esta instrucción no debe ser confundida con SETCOLOR. La instrucción COLOR selecciona uno de los registros de color disponibles para dibujar y pintar. Por ejemplo, en el modo gráfico 7 hay 4 registros de color disponibles para hacer gráficos. Mediante la instrucción COLOR, es posible elegir de entre los 4 registros, para designar puntos o trazar líneas de color en la pantalla. En consecuencia, COLOR es innecesario en los modos de texto, que sólo tienen un registro de color, el cual controla el texto.

El número a continuación de COLOR, K, determina que registro de color se utilizará, dependiendo del modo gráfico en uso.

NUMERACION Y DISPONIBILIDAD DE LOS REGISTROS DE COLOR

NUM.REG. SETCOLOR	VALOR COLOR K EN MODOS GR.		
	3,5,7,15	4,6,14	8*
Ø	1	1	-
1	2	-	1
2	3	-	Ø
3	-	-	-
4	Ø	Ø	-

* En modo gráfico 8, COLOR sólo controla LUMINOSIDAD.

La tabla anterior explica la relación entre COLOR y SETCOLOR.

USOS DE LOS REGISTROS DE COLOR

Modo Gráfico	Ø	Registros de Color			
		1	2	3	4
Ø	-	C*	C,F	-	B
1	C	C	C	C	F,B
2	C	C	C	C	F,B
3	p	p	p	-	p,F,B
4	p	-	-	-	p,F,B
5	p	p	p	-	p,F,B
6	p	-	-	-	p.F.B
7	p	p	p	-	p,F,B
8	-	p*	p,F	-	B
14	p	-	-	-	p,F,B
15	p	p	p	-	p,F,B

NOTA: C= Carácter F= fondo
 p= punto o línea B= borde
 * Solo determina luminosidad

NOTAS:

- Los modos gráficos 12, 13, 14 y 15 son exclusivos de la línea ATARI XL.
- Los modos gráficos Ø, 1, 2, 12 y 13, corresponden a modos de texto.
- Los modos gráficos 9, 10, 11, 12 y 13, están en la sección de Gráficos GTIA.

La instrucción COLOR 3, en los modos gráficos 3, 5 ó 7, selecciona el registro # 2, que puede ser cambiado a cualquier otro color con una instrucción SETCOLOR.

En el modo gráfico Ø, la instrucción COLOR no tiene significado en colores, pero, si a continuación de COLOR se incluye un carácter ASCII, Ud. puede plotear sus gráficos con él.

INSTRUCCION PLOT (PL.) (UBIQUE - PLOTEE)

Su formato es PLOT X,Y. Permite definir un bloque, llamado pixel, mediante la intersección de la columna X y la fila Y. La cantidad de columnas y filas no son las mismas en todos los modos gráficos. Así, la cantidad de pixels disponibles dependerá del modo gráfico.

La instrucción PLOT X,Y en los modos gráficos, es análoga a la instrucción POSITION en los modos de texto. PLOT X, Y ilumina el pixel definido por las coordenadas X,Y. Tal pixel tendrá el color que haya sido definido mediante las instrucciones SETCOLOR y COLOR.

INSTRUCCION DRAWTO X,Y. (DR. X,Y). (DIBUJE HASTA)

Esta instrucción traza una línea, desde el punto fijado por PLOT X,Y hasta el pixel definido por DRAWTO X,Y. El punto inicial también puede haber sido definido por una instrucción DRAWTO anterior.

A continuación, veremos cada modo gráfico detalladamente, y daremos algunos ejemplos que ayuden a comprender mejor la mecánica del empleo de las instrucciones que acabamos de reseñar.

MODO DE GRAFICOS 0

FORMATO: GRAPHICS 0. (GR.0).

Este es el modo normal de texto, y se obtiene por alguna de las siguientes acciones (pantalla azul normal):

- encender el computador
- oprimir [SYSTEM RESET]
- escribir GR.0 [RETURN].

GR.0 cuenta con una pantalla dividida en 40 columnas y 24 filas. El computador se salta las primeras dos columnas y empieza a mostrar el texto, en pantalla, a partir de la tercera columna. Sin embargo, al pasar la información a la impresora, lo hace desde la columna cero. GR.0 es, fundamentalmente, un modo de texto. Tiene sólo un color para TEXTO. Sólo es posible cambiar el color de la pantalla y el del borde. No se definen pixels, y no acepta instrucciones PLOT ni DRAWTO.

Los registros de color, en GR.0 son los siguientes:

Registro de color # 0 : no se usa
Registro de color # 1 : azul claro
Registro de color # 2 : azul oscuro
Registro de color # 3 : no se usa
Registro de color # 4 : negro

El color azul claro coloreado en el registro # 1, controla la luminosidad de los caracteres tales como los de la palabra "READY".

El color azul oscuro del registro # 2, controla el color de la pantalla o "background".

El color negro del registro # 4, controla el color del borde alrededor del "background".

Los registros 0 y 3 no son utilizados en este modo.

INSTRUCCION SETCOLOR EN GR.0.

Como se ha dicho, el computador ATARI tiene registros de color para "guardar" o "almacenar" colores. Los colores específicos almacenados, están determinados por el modo de TEXTO o GRAFICO que se esté usando.

Usando el modo de texto normal, o GR.Ø, podemos variar el color del "background" y el del borde, pero no del texto o "foreground".

Escriba: GR.Ø
y luego: SETCOLOR 1, cualquier #, (Ø a 14)

El registro de color 1, controla la brillantez de los caracteres.

El segundo número de la instrucción puede ser cualquiera. No tiene efecto, sólo hay un color disponible. El color del texto es el mismo que el del fondo.

El tercer número selecciona la brillantez. Ø es el más oscuro, y 14 es el más brillante. (Sólo los números pares tienen influencia. El 3 será igual de brillante que el 2 y el 13 igual que el 12.).

Experimente y observe las diferencias.

Escriba: SE. 4,N,8 N=(Ø, 1, 2,....., 14, 15)

El registro de color 4, controla el borde de la pantalla, por lo que el segundo número de la instrucción selecciona el color del borde.

El tercer número selecciona la brillantez. Ø es el más oscuro y 14 el más brillante.

Experimente y observe las diferencias.

Escriba: SE. 2,6,8
 SE. 1,Ø,8

Qué observa?

El texto ha desaparecido. Sucede que la segunda instrucción controla la brillantez de los caracteres (registro de color # 1) y, siendo el color del texto el mismo que el del fondo y el nivel de brillantez también el mismo, en ambas instrucciones, el texto se confunde con el fondo.

Ensaye con este programa:

```
1Ø PRINT "[ESC]-[CTRL] [CLEAR]": REM LIMPIA PANTALLA
2Ø POKE 752,1: REM HACE DESAPARECER CURSOR
3Ø POSITION 1Ø,11
4Ø PRINT "AHORA, PARPADEA!"
5Ø SETCOLOR 2,12,8: REM BACKGROUND
6Ø FOR R=1 TO 1ØØ
7Ø FOR PARPADO=Ø TO 14 STEP 2
8Ø SE. 1,R,PARPADO
```



```
90 FOR DEM=1 TO 15: NEXT DEM
100 NEXT PARPADO
110 NEXT R
```

Pongamos otro color al borde:

```
85 SE. 4,6,6: REM COLOR DEL BORDE
```

Reemplace la línea 10 por:

```
10 GR.0
```

Resultado: ambas líneas son equivalentes.

Diferencia: GR. Cambia de modo gráfico, borrando todo. [ESC]-[CTRL] [CLEAR] sólo limpia la pantalla, manteniendo el programa en el modo gráfico que haya sido escogido previamente.

INSTRUCCION COLOR

Esta instrucción no tiene uso en GR.0.

MODO DE GRAFICOS 1.

FORMATO: GRAPHICS 1. (GR.1).

El segundo modo de texto es GR.1. Se logra escribiendo GR.1 [RETURN]. Escribalo.

Observará que el texto aparece en una franja en la parte baja de la pantalla, reservada especialmente para este propósito. En esta franja, llamada VENTANA DE TEXTO, caben hasta cuatro líneas de escritura.

En la ventana Ud. puede escribir líneas normales de texto.

Escriba: PRINT " HOLA VENTANA DE TEXTO "

Ahora ensaye lo siguiente y observe atentamente:

PRINT #6; " HOLA PANTALLA "

Bien!, ya sabe como escribir en la ventana y, como escribir en la pantalla!

GR. 1 cuenta con una pantalla dividida en 20 columnas y 20 filas.

A continuación, tenemos los registros de color en GR.1, sus colores normales (default), y el control al que se encuentran asociados.

Registro	Color	Control de
0	naranja	mayúsculas
1	verde	minúsculas
2	azul oscuro	mayúsculas, video inverso
3	rojo	minúsculas, video inverso
4	negro	color del background

Escriba:

- a) PRINT #6; "GRAFICO 1" (en mayúsculas)
- b) PRINT #6; "atari" (en minúsculas)
- c) PRINT #6; "COLORES" (en mayúsculas, video inverso)
- d) PRINT #6; "normales" (en minúsculas, video inverso)

Se obtiene:

- a) GRAFICO 1, en letras de doble ancho, color naranja
- b) ATARI, en letras de doble ancho, color verde
- c) COLORES, en letras de doble ancho, color azul
- d) NORMALES, en letras de doble ancho, color rojo.

USO DE SETCOLOR.

Con la instrucción SETCOLOR, es posible cambiar el color de las palabras mostradas en la pantalla.

Si escribe:		Cambia a:
SE. 0,12,8	GRAFICO 1	Verde
SE. 1,3,8	ATARI	Rojo-Naranja
SE. 2,5,8	COLORES	Púrpura, como la ventana de texto
SE. 3,8,10	NORMALES	Azul
SE. 4,0,8	el fondo	Gris

Escriba:

```

3 GR. 1
5 POS. 5,6: PRINT #6;"GRAFICO 1"
10 POS. 5,8: PRINT #6;"atari"
15 POS. 5,10: PRINT #6;"COLORES" (en video inverso)
20 POS. 5,12: PRINT #6;"normales" (en video inverso)
21 FOR D=1 TO 1000: NEXT D
25 FOR Y=0 TO 15: FOR X=0 TO 15
60 POS. 5,12: PRINT #6;"variables" (en video inverso):
SETCOLOR 3,X,14
65 SETCOLOR 4,Y,0: FOR D=1 TO 250: NEXT D: NEXT X: NEXT Y:
GOTO 25

```

Las instrucciones COLOR, PLOT y DRAWTO, no se utilizan en GR.1. Estas instrucciones son aprovechadas, en toda su potencialidad, en los modos gráficos.

MODO DE GRAFICOS 2.

FORMATO: GRAPHICS 2. (GR.2).

GR.2 es el tercer modo de texto, y es igual a GR.1, con excepción del formato de la pantalla.

Formato de pantalla:

GR.1 - 20 columnas x 20 filas

GR.2 - 20 columnas x 10 filas

Esto significa que, al utilizar el comando PRINT #6 en GR.2, en vez de obtener caracteres de doble ancho y altura normal, como en GR.1, obtendremos caracteres de DOBLE ANCHO Y DOBLE ALTO.

Repita los ejemplos mostrados para =GR.1, utilizando GR.2 en lugar de GR.1.

MODOS DE GRAFICOS 3, 5 Y 7

FORMATO: GRAPHICS 3, 5 Y 7. (GR.3, GR.5, GR.7)

Estos modos de gráficos corresponden a GRAFICOS puros. Los registros de color para estos modos gráficos son los mismos, y son:

REGISTRO	COLOR
Ø	naranja
1	verde claro
2	azul
3	no se usa
4	negro (color normal del fondo)

Estos modos gráficos le permiten:

- Cambiar los colores de los registros de color, utilizando la instrucción SETCOLOR, o
- Utilizar la instrucción COLOR para tomar el color de los registros Ø, 1, 2 o 4, y usarlos en los pixels o líneas que Ud. defina en la pantalla.

Naturalmente, ambas acciones pueden combinarse, permitiéndole elegir el color que pondrá en un determinado registro (con el 2o. número de SETCOLOR), y utilizarlo mediante la instrucción COLOR, ligada a dicho registro de color.

A cada registro de color le corresponde la siguiente instrucción COLOR:

El color del REGISTRO	Se obtiene utilizando
Ø	COLOR 1
1	COLOR 2
2	COLOR 3
4	COLOR Ø

Escriba: GR.5
 COLOR 1
 PLOT 2Ø,1Ø: DRAWTO 6Ø,1Ø
 COLOR 2
 PLOT 2Ø,15: DRAWTO 6Ø,15
 COLOR 3
 PLOT 2Ø,2Ø: DRAWTO 6Ø,2Ø

La primera línea es naranja, la segunda verde, y la tercera azul.

Se han utilizado los colores normales que se encuentran en los registros de color, empleando la instrucción COLOR correspondiente.

Ahora, escriba:

SE. 0,7,4	La 1a. línea se vuelve azul.
SE. 1,0,14	La 2a. línea se torna blanca, y los caracteres de la ventana se hacen más brillantes porque el registro 1, además, controla la luminosidad de los caracteres de la ventana de texto.
SE. 2,12,6	La 3a. línea se vuelve verde, y también la ventana de texto.
SE. 3,6,8	No sucede nada, porque el registro de color #3 no se emplea en este modo gráfico.
SE. 4,4,0	El fondo y el borde se vuelven rojos.

Ensaye y analice el siguiente programa:

```

5 GR.5
10 COLOR 1: PLOT 20,10: DRAWTO 60,10
20 COLOR 2: PLOT 20,15: DRAWTO 60,15
30 COLOR 3: PLOT 20,20: DRAWTO 60,20
40 COLOR 4: FOR D=1 TO 2000: NEXT D
45 FOR X=0 TO 15
50 R=INT (RND(0)*15)+1
52 SETCOLOR 4,R,0
55 FOR Y=0 TO 3: FPR D=1 TO 500: NEXT D
60 SETCOLOR Y,X,4
80 NEXT Y: FOR D=1 TO 250: NEXT D: NEXT X
85 GOTO

```

Como habrá notado, los colores disponibles para los modos de gráfico 3, 5 y 7, en forma simultánea, son cuatro. Tres de ellos para el frente y uno para el fondo.

La diferencia entre estos modos de gráfico está en la resolución de la pantalla, y en la cantidad de memoria que ocupan.

Modo Gráfico	Columnas	Filas
GR.3	40	20
	40	24*
GR.5	80	40
	80	48*
GR. 7	160	80
	160	96*

* sin ventana de texto.

MODOS DE GRAFICOS 4 Y 6

FORMATO: GRAPHICS 4 Y 6. (GR.4, GR.6)

Los registros de color para estos modos gráficos, y los COLOR asociados a ellos, son los siguientes:

Registro

- Ø : color naranja.
- 1 : intensidad de los caracteres en ventana de texto.
- 2 : color azul de la ventana de texto.
- 3 : no se usa.
- 4 : color negro del fondo.

Las instrucciones COLOR, para obtener los colores de los registros, son:

COLOR 1 para el registro Ø
COLOR Ø para el registro 4

La resolución de estos modos gráficos es la siguiente:

Modo Gráfico	Columnas	Filas
GR.4	8Ø	4Ø
	8Ø	48*
GR.6	16Ø	8Ø
	16Ø	96*

* sin ventana de texto.

La diferencia entre estos dos modos gráficos y los modos 5 y 7, está en la cantidad de memoria que utilizan, siendo GR.5 y GR.7 los que ocupan la mayor cantidad, puesto que le ofrecen un mayor número de colores.

Haga el siguiente ejemplo:

```
1Ø GR.6
2Ø SE.Ø,4,8: COLOR 1: PLOT 4Ø,1Ø: DRAWTO 4Ø,6Ø: DRAWTO 41,6Ø:
DRAWTO 41,1Ø
25 FOR D=1 TO 1ØØØ: NEXT D
3Ø SE.2,12,2
35 FOR D=1 TO 1ØØØ: NEXT D
4Ø SE.4,1,Ø: COLOR Ø: PLOT 1ØØ,1Ø: DRAWTO 1ØØ,6Ø
45 FOR D=1 TO 1ØØØ: NEXT D
5Ø SE.4,6,Ø
```


MODO DE GRAFICOS 8

FORMATO: GRAPHICS 8. (GR.8).

Este modo gráfico es el de más alta resolución. Posee 320 columnas y 160 filas, o bien, 320 columnas y 192 filas, sin la ventana de texto.

Los registros de color que le corresponden son:

Registro

- 0 : no se usa.
- 1 : intensidad de los caracteres en la ventana de texto, y de los puntos (o pixels) en la pantalla.
- 2 : color azul. Corresponde al fondo y a la ventana de texto.
- 3 : no se usa.
- 4 : color negro del borde.

Las instrucciones de COLOR que le permiten utilizar los colores de los registros son:

COLOR 1 para el registro 1
COLOR 0 para el registro 2

Copie el siguiente ejemplo:

```
5 GR.8
10 SE.1,4,14: COLOR 1: PL.100,40: DR.100,120: DR.220,120
15 DR.220,40: DR.100,40: FOR D=1 TO 500: NEXT D
20 SE.2,7,0
30 COLOR 0: PL.160,50: DR.160,110: FOR D=1 TO 500:NEXT D
40 SE.4,1,0
```

Experimente sobre esta base y vea que ocurre al cambiar los números, o al eliminar instrucciones.

RESUMEN DE SETCOLOR Y COLOR

Los registros de SETCOLOR pueden utilizarse para controlar diversas cosas:

- a) El color del borde o el del fondo.
- b) El color y/o la brillantez de los puntos, o líneas, que Ud. dibuja en la pantalla.
- c) La brillantez (luminosidad) de los caracteres en la pantalla.

La instrucción COLOR se utiliza (desde GR.3 en adelante) para indicarle al computador que Ud. desea dibujar una línea o un punto, con un color que se encuentra almacenado en un registro de color determinado. Asegúrese que el número del COLOR que eligió sea el que corresponde al registro utilizado en la instrucción SETCOLOR.

GRAFICOS GTIA

El computador ATARI posee, además, otros tres modos de gráfico y dos de texto (sólo línea XL), controlados por un chip llamado GTIA. ("chip" es el nombre comúnmente dado a un circuito integrado).

MODOS DE GRAFICOS 9, 10 y 11

Los modos de gráfico están numerados del 9 al 11, y cada una presenta hasta 16 colores en la pantalla, en forma simultánea. Todos tienen una resolución de 80 puntos, en sentido horizontal, y 192 puntos, en sentido vertical. Es decir que los pixels son cuatro veces más anchos que altos. Las variantes entre los tres modos se encuentran explicadas a continuación.

Modo	Colores	Intensidades	Bytes
9	1	16	8138
10	9	9	8138
11	16	1	8138

Para el modo de gráfico número 9 hay UN sólo registro de color disponible, el 4, que controla el color del fondo. La instrucción COLOR selecciona las distintas intensidades, de este color del fondo, con las que Ud. puede dibujar.

Haga el siguiente ejemplo:

```

10 GR.9
20 SE.4,11,0
25 FOR X=1 TO 15
30 COLOR X: PLOT (X+10): DRAWTO (70-X),180
40 FOR D=1 TO 200: NEXT D: NEXT X
50 GOTO 25

```

En el modo de gráfico 10, hay NUEVE registros de color, que se seleccionan mediante una instrucción POKE, seguida del número del registro y del número del color, multiplicado por 16, más la intensidad que desea darle.

Los registros de color están numerados del 704 al 712, inclusive. El registro 704 controla el color del fondo, y los otros guardan los ocho colores restantes. Las instrucciones COLOR, que seleccionan estos registros, van del 0 al 8, respectivamente.

Ejemplo de como asignar un color a un registro dado:
 POKE 708,12*16+8 (reemplaza a la instrucción SETCOLOR).

En el modo de gráfico 11, Ud. tiene UN solo registro de color disponible el 4, que controla la intensidad de los colores que utilizará. La instrucción COLOR selecciona los diferentes colores con los que Ud. puede dibujar:

Haga el siguiente ejemplo:

```
10 GR.11
20 SE.4,0,10
25 FOR X=1 TO 15
30 COLOR X: PLOT (X+10): DRAWTO (70-X), 180
40 NEXT X
50 GOTO 25
```

MODOS DE GRAFICOS 12 y 13

Los modos de texto se seleccionan con GR.12 y GR.13 (estos son exclusivos para la línea ATARI XL). Se utilizan principalmente para efectos especiales, como escenarios para juegos, conjuntos de caracteres redefinidos y otros.

Otra particularidad es que las letras son multicolores.

La siguiente tabla explica las características de estos modos de texto.

Modo	Columnas	Filas	Bytes
12	40	20 6 24	992
13	40	10 6 12	674

Estos modos de texto, al igual que GR.0, le permiten utilizar los caracteres gráficos de texto, que se obtienen al oprimir la tecla de [CONTROL] y las letras, en forma simultánea. Esto es, trazar formas simples en los modos de texto. En este manual se incluye una plantilla con los caracteres gráficos y su tecla correspondiente, al final del libro.

MODOS DE GRAFICOS 14 y 15 (línea XL)

Los computadores ATARI, pertenecientes a la línea XL (600 XL, 800XL, etc.), poseen cuatro modos gráficos adicionales. Dos de ellos son los modos de texto 12 y 13, que ya fueron explicados en la sección de gráficos GTIA, y los otros dos son los modos gráficos de alta resolución 14 y 15, que serán tratados a continuación.

Los modos 14 y 15 son modos de alta resolución, y la única diferencia entre ellos es la cantidad de colores que pueden presentar simultáneamente en la pantalla, y la cantidad de memoria que ocupan.

En la siguiente tabla, presentamos las características de estos modos.

Modo	Columnas	Filas	Colores	Bytes
14	160	192	2	4296
15	160	192	4	8138

Cabe destacar que el modo de gráfico 15, es el modo de 4 colores de más alta resolución que posee este computador.

Existen medios para obtener estos modos en los computadores ATARI 400 y 800, pero se requiere un proceso más complicado, materia que no es del alcance de este libro.

NOTA 1: Con el impresor ATARI 1020 a colores, es posible, imprimir la mayoría de sus gráficos en papel, a cuatro colores. (verde, rojo, azul y negro).

NOTA 2: La tableta de gráficos ATARI, le permite dibujar, con facilidad, cualquier cosa en la pantalla, sin necesidad de un programa especial. Se puede pintar a gusto con un par de movimientos del lápiz, y Ud. puede almacenar sus dibujos (de Biología, Geometría, Física, o Artes Plásticas) en cassettes o diskettes. También puede imprimirlos.

SONIDO

El computador ATARI posee cuatro sintetizadores de sonido independientes. Cada uno puede producir 255 frecuencias distintas, a las cuales se les puede modificar el volumen y la distorsión a voluntad. Esto le permite interpretar música y crear efectos especiales.

```

Escriba : 10 Z = 0 TO 255
          20 SOUND 0,Z,10,10
          30 NEXT Z

```

Ahora escriba RUN. No pasa nada? Aumente el volumen de su televisor.

```

Agregue: 40 FOR Z=255 TO 0 STEP -1
          50 SOUND 0,Z,10,10
          60 NEXT Z
          70 GOTO 10

```

Una forma de crear sonidos en el computador ATARI es por medio del comando SOUND, cuyo formato es el siguiente:

SOUND R,N,D,V

R - 0 a 3. Es el registro de sonido. Existen cuatro registros de sonido, también llamados voces.

N - 0 a 255. Es el sonido en sí mismo, podría definirse como 'nota' que genera el computador. Cada valor representa una frecuencia distinta.

D - 0 a 14 (sólo los pares). Representa el nivel de distorsión del sonido. El 10 da un sonido puro, sin distorsión.

V - 0 a 15. Es el volumen del sonido.

También puede abreviarse: SO. R,N,D,V

Ud. puede generar sonidos totalmente distintos para cada registro.

Con el computador en modo directo, escriba:

```

SOUND 0.50,10,6
SOUND 1,100,10,6
SOUND 1,150,10,6

```


SOUND 3,200,10,6

Aquí tiene un programa de ejemplo:

```
10 GR. 7+16: SE.4,0,0: SE.1,0,4
20 COLOR 2
30 PL.2,2: DR.30,34: DR.78,40: DR.100,57: DR.110,50:
DR.140,76
40 FOR SON=1 TO 255
50 SO.0,SON,8,10
60 IF SON=8 THEN SETCOLOR 1,0,14
70 NEXT SON
80 SE.1,0,0
90 FOR D=1 TO 200: NEXT D
100 GOTO 10
```


VALORES DE SONIDO PARA LAS NOTAS MUSICALES

NOTAS ALTAS	DO	29
	SI	31
	LA#	33
	LA	35
	SOL#	37
	SOL	40
	FA#	42
	FA	45
	MI	47
	RE#	50
	RE	53
	DO#	57
	DO	60
	SI	64
	LA#	68
	LA	72
	SOL#	76
	SOL	81
	FA#	85
DO CENTRAL	FA	91
	MI	96
	RE#	102
	RE	108
	DO#	114
	DO	121
	SI	128
	LA#	136
	LA	144
	SOL#	153
	SOL	162
	FA#	173
	FA	182
	MI	193
	RE#	204
	RE	217
	DO#	230
	DO	243
NOTAS BAJAS		

EL SIMBOLO # SIGNIFICA " SOSTENIDO ".

EJERCICIO

CIRCULOS MULTICOLORES.

```

10 GR.5: C=1
20 PRINT "TAMAÑO";: INPUT P
30 FOR A=2 TO P STEP 2
40 FOR N=P-A TO P+A
50 COLOR C: C=C+1: IF C=4 THEN C=1
60 PL.N,P+SQR(A^2-(N-P)^2)
70 PL.N,P-SQR(A^2-(N-P)^2)
80 NEXT N
90 NEXT A
100 END

```

STAR WARS (música)

```

10 GR.0: SE.2,13,0: POS.8,10: PRINT "La Guerra de las
Galaxias.": POS.16,12: PRINT "(música)"
20 FOR LARGO=1 TO 20
30 READ TONO,DURACION
40 SOUND 0, TONO,10,10: DURACION=DURACION*5
50 FOR D=1 TO DURACION: NEXT D
60 SOUND 0,0,0,0
70 NEXT LARGO
80 RESTORE: GOTO 20
1000
1007,10,207,10,207,10,162,60,108,60,121,10,128,10,144,10,81,
60,108,30
1010 DATA 121,10,128,10,121,10,144,30,0,30
1005 DATA 121,10,128,10,144,10,81,60,108,30

```

DATA

INSTRUCCION PEEK E INSTRUCCION POKE

El lenguaje Basic posee los medios para 'MIRAR' el contenido de una dirección de memoria y alterarlo.

Cada dirección de memoria almacena un Byte, de los 65535 que puede poseer el computador ATARI. Piense en cada ubicación o dirección de memoria como en una casilla de correos, numerada del 0 al 65535.

La instrucción PEEK permite inspeccionar el contenido de cada dirección o ubicación de memoria, y su formato es el siguiente:

PRINT PEEK (710)

(imprime el contenido de la casilla de memoria número 710)

Al hacer uso de: PEEK(xxxxx), el número de la casilla deberá especificarse entre paréntesis.

Ejemplo: 10 A=PEEK(54279)

La instrucción POKE funciona en forma similar, sólo que sirve, para alterar el valor almacenado en una casilla de memoria. Para ello, debe especificar el número de la casilla, seguida del nuevo valor para almacenar en esa ubicación.

El formato de esta instrucción es el siguiente:

POKE 755,4

(almacena el número 4 en la casilla de memoria 755)

Al hacer uso de POKE xxxxx,yyy , la dirección estará separada del valor a almacenar por una coma.

Ejemplo: 10 POKE 752,1

NOTA: NO ES POSIBLE ALMACENAR EN UNA CASILLA VALORES MAYORES QUE 255. EL INTENTARLO PROVOCARA UN ERROR-3, ASIMISMO LA INSTRUCCION PEEK NO RETORNARA UN VALOR MAYOR QUE 255.

A continuación presentamos algunos ejemplos del poder de estos dos comandos.

10 GR.2

20 POS.3,5: POKE 755,0: PRINT #6;"ARRIBA"


```

30 FOR X=1 TO 250: NEXT X
40 POS.3,5: POKE 755,4: PRINT #6;"abajo"
50 FOR X=1 TO 250: NEXT X
60 IF PEEK(53279)=6 THEN END
70 GOTO 20
80 REM --
90 REM
100 REM PRESIONA [START] PARA SALIR DE ESTE PROGRAMA

```

```

10 FOR X=0 TO 255: POKE 710,X: POKE 712,255-X: POKE 709,X/2:
NEXT X: GOTO 10

```

```

10 POKE 559,0: REM PRESIONE [RESET] PARA SALIR DE ESTE
PROGRAMA
20 POKE 53279,8: GOTO 20

```

```

10 REM EN LA LINEA DE COMPUTADORES
20 REM 600XL y 800 XL EL COMANDO
30 REM POKE 756,204 PERMITE EL
40 REM ACCESO AL SET DE
50 REM CARACTERES INTERNACIONALES
60 POKE 756,225
70 PRINT"A,B,C,.....,X,Y,Z"
80 FOR D=1 TO 500: NEXT D
90 POKE 756,204
100 PRINT "[CTRL]A, [CTRL]B, [CTRL],C,..... [CTRL]X, [CTRL]Y,
[CTRL]Z,"
110 FOR D=1 TO 500: NEXT D
120 GOTO 60

```


TABLA RESUMEN DE MODOS GRAFICOS (0-8, 14, 15)

MODO GRAFICO	USO	COLORES DISPONIBLES	RESOLUCION COL. x FIL.	REGISTRO DE PUNTO	REGISTRO DE FONDO	REGISTRO DE BORDE	MEMORIA OCUPADA
0	TEXTTO	1 COL 2 LUM.	20 x 24	1	2	4	992
1	TEXTTO	5	20 x 20 20 x 24*	0,1,2,3	4	4	672
2	TEXTTO	5	20 x 10 20 x 12*	0,1,2,3	4	4	420
3		4	40 x 20 40 x 24*	0,1,2	4	4	432
5	GRAFICOS	4	80 x 40 80 x 48*	0,1,2	4	4	1176
7		4	160 x 80 160 x 96*	0,1,2	4	4	4200
15		4	160 x 160	0,1,2	4	4	8138
4		2	80 x 40 80 x 48*	0	4	4	696
6	GRAFICOS	2	160 x 80 160 x 96*	0	4	4	2184
14		2	160 x 160 160 x 192*	0	4	4	4296
8	GRAFICOS	1 COL 2 LUM.	320 x 160 320 x 192*	1	2	4	8138

* - sin ventana de texto

NOTA: LOS MODOS DE GRAFICO 9, 10, 11, 12 Y 13, ESTAN EXPLICADOS EN LA SECCION DE GRAFICOS GTIA.
LOS MODOS 12, 13, 14 Y 15, SON EXCLUSIVOS DE LA LINEA ATARI-XL (600-XL, 800-XL, ETC.)

CODIGOS DE ERROR

CODIGO DE ERROR	MENSAJE DE ERROR CORRESPONDIENTE
2	Memoria insuficiente
3	Error de Valor Numérico
4	Demasiadas Variables
5	Largo de Variable alfanumérica Excedido
6	Falta de Datos (DATA)
7	Número mayor que 32767 (Número de Línea)
8	Error en INPUT
9	Error en DIMensión de Variable
10	Expresión Matemática demasiado Compleja
11	Superación de la capacidad de manejo de números muy grandes o muy pequeños.
12	Línea inexistente
13	NEXT sin FOR
14	Línea demasiado Larga
15	Línea GOSUB o FOR suprimida
16	RETURN sin GOSUB
17	Error de Sintaxis
18	Variable Alfanumérica No Válida
19	Programa demasiado Largo para ser Cargado
20	Número de Canal 0 o mayor que 7 (OPEN)
21	Error de Carga de un Programa
128	Operación de ENTRADA o SALIDA interrumpida con BREAK
129	Canal abierto previamente
130	Aparato Inexistente
131	Intento de Entrada por un canal de Salida
132	Error de Sintaxis en un comando XIO
133	Canal No Abierto previamente
134	Canal de Salida Desconocido
135	Intento de Salida por un canal de Entrada
136	Fin de Archivo
137	Fichero Truncado
138	Aparato No Responde
139	Aparato No Funciona o No puede ejecutar la instrucción
140	Error en la comunicación entre el aparato y el computador
141	Cursor Fuera de Pantalla
142	Inconsistencia en la comunicación Aparato/Computador
143	Error en la Grabación o Lectura desde Cassette o Diskette.

CODIGOS DE ERROR (Cont.)

CODIGO DE ERROR	MENSAJE DE ERROR CORRESPONDIENTE
144	Intento de Escritura en un Diskette Protegido.
145	Error de Modo de Gráfico (GRAPHICS) o Verificación.
146	Función No Implementada
147	Memoria Insuficiente para Modo Gráfico
160	Número de Unidad de Disco Desconocido
161	Demasiados Archivos Abiertos
162	Diskette Lleno
163	Error Irrecuperable del Sistema
164	Número de Archivo No Corresponde
165	Error en el Nombre del Archivo
166	Error en la Instrucción POINT
167	Archivo Protegido
168	Comando XIO Desconocido
169	Directorio del Diskette Lleno
170	Archivo Inexistente
171	Instrucción POINT No Válida
172	Intento de Añadir Datos a un Archivo con DOS distinto del original.
173	Sectores Malos al momento de Formatear.

CODIGOS DE ERROR (Cont.)

CODIGO DE ERROR	MENSAJE DE ERROR CORRESPONDIENTE
144	Intento de Escritura en un Diskette Protegido.
145	Error de Modo de Gráfico (GRAPHICS) o Verificación.
146	Función No Implementada
147	Memoria Insuficiente para Modo Gráfico
160	Número de Unidad de Disco Desconocido
161	Demasiados Archivos Abiertos
162	Diskette Lleno
163	Error Irrecuperable del Sistema
164	Número de Archivo No Corresponde
165	Error en el Nombre del Archivo
166	Error en la Instrucción POINT
167	Archivo Protegido
168	Comando XIO Desconocido
169	Directorio del Diskette Lleno
170	Archivo Inexistente
171	Instrucción POINT No Válida
172	Intento de Añadir Datos a un Archivo con DOS distinto del original.
173	Sectores Malos al momento de Formatear.

CONTROL GRAPHICS KEYBOARD

